

24^a

ИНСТИТУТ ЯДЕРНОЙ ФИЗИКИ СО АН СССР




А.Г. Грозин

РЕШЕНИЕ ФИЗИЧЕСКИХ ЗАДАЧ
НА ЯЗЫКЕ REDUCE.

- 3. Нелинейные волны на воде
- 4. Вычисление тензора кривизны
- 5. Сложение моментов в квантовой механике

ПРЕПРИНТ 88-136



НОВОСИБИРСК

Решение физических задач на языке REDUCE.

3. Нелинейные волны на воде
4. Вычисление тензора кривизны
5. Сложение моментов в квантовой механике

А.Г. Грозин

Институт ядерной физики
630090, Новосибирск 90, СССР

АННОТАЦИЯ

Настоящий препринт представляет собой вторую часть пособия по применению языка аналитических вычислений REDUCE для решения физических задач. В нем приведено много примеров, которые могут быть полезны для конструирования программ решения физических задач самой различной природы. В этой части рассмотрена задача о нелинейных волнах на воде (п.3), вычисление тензора кривизны (п.4) и сложение моментов в квантовой механике (п.5).

Solving physical problems with REDUCE.

3. Nonlinear water waves
4. Calculation of the curvature tensor
5. Angular momentum addition

A.G. Grozin

Institute of Nuclear Physics
630090, Novosibirsk, USSR

АБСТРАКТ

This preprint is the second part of the problem book on using REDUCE in physics. It contains many examples useful for the construction of programs for solving physical problems of very different nature. This part contains the problem of nonlinear water waves (sect. 3), the calculation of the curvature tensor (sect. 4) and angular momentum addition (sect. 5).

© Институт ядерной физики СО АН СССР

3. НЕЛИНЕЙНЫЕ ВОЛНЫ НА ВОДЕ

Введение. Следующий наш пример относится к гидродинамике. Речь пойдет о волнах на поверхности несжимаемой идеальной жидкости в поле силы тяжести. В этом случае малым параметром является отношение амплитуды a к длине волны λ , или крутизна скатов волны.

В линейном приближении волны на воде рассмотрены, например, в §12 учебника [1]. Нелинейные поправки обсуждаются в главе 5 книги [2]. В ее первом параграфе описано решение в виде бегущей волны в случае бесконечной глубины с точностью до членов $\sim a^3$, и в случае конечной до a^2 . В препринте [3] описано решение задачи на языке REDUCE с точностью до членов $\sim a^3$ при конечной глубине. Описанная там программа требует дополнений в каждом порядке теории возмущений. В этом параграфе мы построим программу, применимую в любом порядке (он входит как параметр), и получим результаты, эквивалентные [3], а для бесконечной глубины продвинемся до членов $\sim a^5$.

Теория. Уравнение движения идеальной жидкости (уравнение Эйлера) имеет вид

$$\rho \frac{d\vec{v}}{dt} = -\vec{\nabla} p + \rho \vec{g}. \quad (1)$$

Здесь $d\vec{v}$ — изменение скорости данной жидкой частицы, $d\vec{v} = \frac{\partial \vec{v}}{\partial t} dt + d\vec{r} \cdot \vec{\nabla} \vec{v}$, откуда ускорение $\frac{d\vec{v}}{dt} = \frac{\partial \vec{v}}{\partial t} + \vec{v} \cdot \vec{\nabla} \vec{v}$; $-\vec{\nabla} p -$

сила давления, а $\rho \vec{g}$ — сила тяжести, действующая на единицу объема жидкости. Условие несжимаемости имеет вид

$$\vec{\nabla} \cdot \vec{v} = 0. \quad (2)$$

Член $\vec{v} \cdot \vec{\nabla} \vec{v}$ в уравнении Эйлера можно переписать в другом виде. Именно, рассмотрим выражение $\vec{v} \times (\vec{\nabla} \times \vec{v})$, и раскроем его по формуле $\vec{a} \times (\vec{b} \times \vec{c}) = \vec{b} \vec{a} \cdot \vec{c} - \vec{c} \vec{a} \cdot \vec{b}$. Чтобы не запутаться, на которое \vec{v} действует $\vec{\nabla}$, напомним это в виде $\vec{v}' \times (\vec{\nabla} \times \vec{v})$, и будем считать, что $\vec{\nabla}$ действует только на \vec{v} , но не на \vec{v}' , где бы она не стояла. Получим $\vec{\nabla} \vec{v} \cdot \vec{v}' - \vec{v} \vec{\nabla} \cdot \vec{v}'$; первый член равен $\vec{\nabla} \vec{v}^2/2$, а второй $\vec{v} \cdot \vec{\nabla} \vec{v}$. Отсюда имеем $\vec{v} \cdot \vec{\nabla} \vec{v} = \vec{\nabla} \vec{v}^2/2 - \vec{v} \times (\vec{\nabla} \times \vec{v})$.

Как известно, в идеальной жидкости циркуляция скорости по каждому жидкому контуру сохраняется. Поэтому, если движение жидкости сначала было безвихревым (например, жидкость покоилась), то оно всегда останется безвихревым, что бы с ней ни делали. Будем искать решение для нашей волны в безвихревом виде, потому что волну можно возбудить на покоящейся жидкости действием потенциальных сил. Для безвихревого движения $\vec{\nabla} \times \vec{v} = 0$, и \vec{v} можно записать в виде $\vec{\nabla} \phi$, где ϕ называется потенциалом скорости. Уравнение Эйлера при этом упрощается (с учетом только что выведенной формулы для $\vec{v} \cdot \vec{\nabla} \vec{v}$):

$$\frac{\partial \vec{\nabla} \phi}{\partial t} + \vec{\nabla} \frac{v^2}{2} = -\frac{\vec{\nabla} p}{\rho} + \vec{g},$$

или

$$\vec{\nabla} \left(\frac{\partial \phi}{\partial t} + \frac{v^2}{2} + \frac{p}{\rho} - \vec{g} \cdot \vec{r} \right) = 0,$$

Значит, выражение в скобках постоянно в пространстве. К ϕ всегда можно прибавить произвольную функцию времени, и добиться, чтобы оно было равно нулю. Поэтому для безвихревых движений уравнение Эйлера приводит к интегралу Бернулли:

$$\frac{\partial \phi}{\partial t} + \frac{v^2}{2} + \frac{p}{\rho} - \vec{g} \cdot \vec{r} = 0. \quad (3)$$

А условие несжимаемости (2) дает уравнение Лапласа для ϕ :

$$\vec{\nabla}^2 \phi = 0. \quad (4)$$

Направим ось z вертикально вверх, а ось x вдоль направления распространения волны, тогда $\phi = \phi(t, x, z)$. Обозначим поверх-

ность жидкости через $z_0(t, x)$. Уравнение (3) определяет давление в любой точке этой жидкости. В частности, на поверхности оно должно быть равно внешнему давлению, которое всегда можно положить равным нулю (если это не так, то добавлением к ϕ линейной функции времени можно сделать правую часть (3) равной $p_{\text{внеш}}/\rho$). Поэтому мы имеем граничное условие

$$\left(\frac{\partial \phi}{\partial t} + \frac{1}{2} \left(\frac{\partial \phi}{\partial x} \right)^2 + \frac{1}{2} \left(\frac{\partial \phi}{\partial z} \right)^2 + gz \right)_{z=z_0(t, x)} = 0. \quad (5)$$

Второе граничное условие состоит в том, что нормальная компонента скорости воды на поверхности равна скорости перемещения этой поверхности (рис. 1). На рисунке $\vec{l} = (1, \frac{\partial z_0}{\partial x})$ есть век-




Рис. 1.

тор, касательный к поверхности, следовательно, $\vec{n} = (-\frac{\partial z_0}{\partial x}, 1)$ есть ненормированный вектор, перпендикулярный поверхности. Перемещение поверхности жидкости по вертикали за промежуток времени dt дается вектором $d\vec{a} = (0, \frac{\partial z_0}{\partial t} dt)$, а по нормали — его проекцией на \vec{n} : $d\vec{a} \cdot \vec{n} / |\vec{n}|$. Это перемещение, деленное на dt , должно быть равно нормальной компоненте скорости $\vec{\nabla} \phi \cdot \vec{n} / |\vec{n}|$.

Отсюда имеем второе граничное условие

$$\frac{\partial z_0}{\partial t} = \left(\frac{\partial \varphi}{\partial z} - \frac{\partial \varphi}{\partial x} \frac{\partial z_0}{\partial x} \right)_{z=z_0(t, x)} \quad (6)$$

Величины $\varphi(t, x, z)$ и $z_0(t, x)$ пропорциональны амплитуде волны a . В нулевом приближении мы отбрасываем в граничных условиях (5) и (6) все квадратичные по a члены. С той же точностью, в производных $\frac{\partial \varphi}{\partial t}$ и $\frac{\partial \varphi}{\partial z}$ можно полагать $z=0$ вместо $z=z_0(t, x)$. Тогда

$$\left(\frac{\partial \varphi}{\partial t} \right)_{z=0} = g z_0, \quad \frac{\partial z_0}{\partial t} = \left(\frac{\partial \varphi}{\partial z} \right)_{z=0}$$

Ищем решение в виде бегущей волны: $\varphi_{z=0} = a \cos(kx - \omega t)$. Из уравнения (4) следует, что зависимость от z должна иметь вид $\exp(\pm kz)$. В случае бесконечной глубины мы должны выбрать $\varphi = a \cos(kx - \omega t) \exp(kz)$. Сначала рассмотрим именно этот случай; для конечной глубины h надо заменить $\exp(kz)$ на $\operatorname{ch} k(z+h)$, чтобы при $z=-h$ иметь $v_z = \frac{\partial \varphi}{\partial z} = 0$. В результате

получаем $z_0 = -\frac{1}{g} \frac{\partial \varphi}{\partial t}$, и уравнения удовлетворяются при $\omega^2 = gk$.

В этом приближении все уравнения линейны, и любая линейная комбинация решений тоже является решением. Таким образом, в линейном случае мы имеем полное решение задачи: любое волновое движение жидкости разлагается по волнам вида $\varphi = a \cos(kx - \omega t) \exp(kz)$ с $\omega^2 = gk$.

Теперь перейдем к рассмотрению высших членов разложения по амплитуде. При этом возникают новые явления: искажается форма волны, возникает взаимодействие волн между собой. Разумеется, мы не можем рассчитывать на полное решение задачи — описание нелинейного волнового движения жидкости общего вида. Это невозможно уже в случае 2 степеней свободы, и тем более невозможно для нелинейных волн, имеющих бесконечное число степеней свободы. Поэтому поставим перед собой ограниченную задачу: будем искать волну, периодическую по x и распространяющуюся без искажений как единое целое. При малых амплитудах такая волна переходит в определенную линейную волну, решение для которой было найдено в предыдущем абзаце. С ростом a появляется, как и для осциллятора, зависимость $\omega(a)$.

Рассмотрим сначала случай бесконечно глубокой воды. Задача формулируется в виде уравнения (4) с граничными условиями (5), (6). В этой формулировке не появляются величины с размерностью массы. Поэтому мы можем выбрать 2 величины с разными размерностями за единицу, зафиксировав тем самым единицы измерения длины и времени. Выберем $g=1$ и $k=1$, тогда $\omega_0=1$. Параметр a является безразмерным, и характеризует малость амплитуды волны. Нулевое приближение имеет вид $\varphi = a \cos \xi \exp(z)$, где $\xi = x - vt$; следующие приближения содержат гармоники $\cos j \xi \exp(jz)$ и $\sin j \xi \exp(jz)$, тогда уравнение (4) удовлетворяется автоматически.

Докажем, что решение можно искать в виде

$$\varphi = a \cos \xi e^z + \sum_{i=2}^{\infty} a^i x \begin{cases} i \text{ — чет:} & \sum_{j=0, j \text{ — чет}}^i b_{ij} \sin j \xi e^{jz} \\ i \text{ — нечет:} & \sum_{j=3, j \text{ — нечет}}^i b_{ij} \cos j \xi e^{jz} \end{cases} \quad (7)$$

То, что в данном порядке i максимальный номер гармоники j равен i , и то, что при четном i все j четные, а при нечетном нечетные, объясняется так же, как и для нелинейного осциллятора. Точно так же, гармоника с $j=1$ полностью содержится в невозмущенном решении. Новым моментом является расположение синусов и косинусов, именно с ним и надо разобраться.

Назовем нечетными величины, в которых, как в (7), при нечетных степенях a стоят косинусы, а при четных синусы. Наоборот, величины, в которых при нечетных степенях a стоят синусы, а при четных косинусы, назовем четными. Используя тот факт, что произведение косинуса на косинус и синуса на синус линейно выражается через косинусы, а косинуса на синус через синусы, легко проверить, что произведения двух четных или двух нечетных величин являются четными, а четной на нечетную нечетным. Операторы $\partial/\partial t$ и $\partial/\partial x$ меняют местами косинусы и синусы и поэтому являются нечетными, а $\partial/\partial z$ является четным. Предположим, что φ является нечетной величиной, а z_0 — четной. В граничных условиях (5), (6) φ и его производные входят при $z=z_0(t, x)$. Они могут быть разложены в ряд Тэйлора по z_0 , а поскольку z_0 — четная величина, их четность при этом не меняется. Тогда легко убедиться, что в граничном условии (5) все члены четны, а в (6) нечетны. Это и доказывает самосогласованность нашего предполо-

жения, т. е. правильность предложенной формы решения (7).

Алгоритм. Итак, мы ищем решение в виде (7). В граничных условиях (5), (6) φ и его производные входят при малых $z = z_0(t, x) \sim a$, поэтому все экспоненты в (7) можно сразу разложить в ряд Тэйлора. Далее, формулу (5) можно использовать для того, чтобы выразить $z_0(t, x)$ через те же параметры b_{ij} . А именно, перепишем ее в виде

$$z_0(t, x) = - \left[\frac{\partial \varphi}{\partial t} + \frac{1}{2} \left(\frac{\partial \varphi}{\partial x} \right)^2 + \frac{1}{2} \left(\frac{\partial \varphi}{\partial z} \right)^2 \right]_{z=z_0(t, x)} \quad (8)$$

Итерируя это уравнение (т. е. подставляя на каждом новом шаге в правую часть для $z_0(t, x)$ выражение, полученное на предыдущем шаге), можно получить $z_0(t, x)$ с любой желаемой точностью по a .

После этого уравнение (6) можно использовать для нахождения неопределенных коэффициентов. В первую очередь выделяем в нем первую гармонику, и получим, что $(v^2 - 1)a \cos \xi$ равно первой гармонике в нелинейном члене. В ней также содержится v , поэтому уравнение опять нужно решать итерациями. В конце концов мы получим выражение для v через коэффициенты b_{ij} . В отличие от осциллятора, теперь v входит и линейно, поэтому приходится извлекать корень из ряда. Далее уравнение можно решать последовательно, приравнявая члены при a^2, a^3, \dots . Когда рассматриваем порядок a^i , получаем выражение для b_{ij} через ранее вычисленные коэффициенты $b_{i'j'}$ с $i' < i$.

Программа. Сначала присвоим потенциалу скорости FF общий вид решения (7). Переменная Y обозначает ξ ; ее производные по X и T заданы подстановками. Далее разложим FF и его производную FZ в ряд Тэйлора по Z. Имеющаяся версия REDUCEa рассматривает $E**(N*Z)$ как $(E**Z)**N$ (N — целое). Поэтому, чтобы не заниматься лишней работой по возведению рядов в степень, сначала заменим $E**Z$ на свою функцию EX(Z), а затем $EХ(Z)**N$ на EX(N*Z). Далее, в членах с разными степенями A функцию EX нужно разлагать до разных порядков. Чтобы запомнить, какой экземпляр EX куда разлагать, введем в эту функцию дополнительный параметр, равный требуемому порядку разложения. На этом примере можно заметить, что, однажды описав функцию командой OPERATOR, ее можно далее использовать с различным числом параметров — эта команда фактически определяет семейство функций с одинаковым именем и с 0, 1, 2, ... аргу-

ментами. Наконец, производится собственно разложение.

Затем применим формулу (8) для уровня воды ZZ. Далее нужно подставлять ZZ вместо Z в ZZ до тех пор, пока ZZ не перестанет зависеть от Z. Для проверки этого используется условие FREEOF(ZZ,Z), которое истинно, если ZZ не содержит Z. Разумеется, ZZ не будет содержать Z через конечное число оборотов цикла только в силу того, что все члены, превышающие заданный порядок малости, отбрасываются. Чтобы не делать лишней работы, желательно считать Z величиной 1-го порядка малости (как оно и есть на самом деле). Но тогда текущая версия REDUCEa не позволит подставлять выражение вместо Z при помощи функции SUB. Поэтому подстановка производится в два приема. Сначала делаем копию ZZ, называемую ZZ1, в которой на месте Z стоит свободная переменная Z1. Это делается при выключенном порядке малости Z. Потом его включаем, и подставляем в ZZ1 вместо Z1 выражение ZZ (содержащее Z!).

Наконец, выражение для уровня воды ZZ через коэффициенты B(I,J) готово. Теперь строим выражение Q, которое должно быть равно нулю в силу второго граничного условия (6). В этом выражении заменим $V**2$ на $1+W$, где поправка W имеет 2-й порядок малости. Поскольку V встречается и в 1-й степени, мы заменяем его на разложение $(1+W)**(1/2)$ до нужного члена. Теперь надо выделить из Q первую гармонику и, приравняв ее нулю, получить для поправки W к квадрату частоты (или фазовой скорости) $V**2$ выражение через коэффициенты B(I,J). Мы знаем, что коэффициент при $\cos(Y)$ в Q (обозначим его R) имеет вид $A*W + R'$, где R' начинается с членов $A**3$ (выражение $A*W$ получается в линейном приближении). Поэтому получаем для поправки $W = -R'/A$. К сожалению, R' содержит W, поэтому это уравнение надо решать итерациями. Строим выражение $WW = -R'/A = W - R/A$, и заменяем в нем W на WW до тех пор, пока WW не перестанет зависеть от W. Эти итерации проводятся точно так же, как для ZZ (только W имеет 2-й порядок малости). То, что получилось, и есть выражение для W через B(I,J); оно начинается с членов $A**2$.

Последний шаг решения — это нахождение коэффициентов B(I,J) из уравнения $Q=0$ (где из Q удалена первая гармоника). Мы выделяем из Q члены с различными степенями A (функцией COEFF), и рассматриваем их поочередно. Для каждой степени $A**I$, разлагаем коэффициент в Q при этой степени на гармоники — коэффициенты при $\cos(J*Y)$. Такой коэффициент линейно

содержит $B(I,J)$ (он содержит также ранее вычисленные $B(I',J')$ с $I' < I$). Приравнивая его нулю, мы находим $B(I,J)$. Это делается по простой формуле: если F — линейная функция X , то решение уравнения $F=0$ есть $X = -\text{SUB}(X=0,F)/\text{DF}(F,X)$.

Присваивания $\text{FF}:=\text{FF}$; $\text{ZZ}:=\text{ZZ}$; $\text{W}:=1+\text{W}$; дают окончательные выражения для потенциала скорости, уровня воды и квадрата частоты (или фазовой скорости, что одно и то же в силу $k=1$), в которые подставлены найденные коэффициенты $B(I,J)$. Мы специально использовали другую переменную F для разложения FF в ряд по Z , чтобы теперь иметь возможность распечатать неразложенный потенциал скорости.

Наконец, еще одно обстоятельство. Первоначально программа выглядела в точности так, как здесь описано. Просчитав ее для 2-го порядка теории возмущений, я с удивлением обнаружил, что обе поправки к ϕ равны нулю. Это значит, что в общем виде решения (7) нет членов a^2 и a^3 , т. е. сумма по i начинается с 4 (в случае бесконечно глубокой воды). Это счастливое обстоятельство позволяет без слишком больших затрат продвинуться еще на 2 порядка. Здесь приводится распечатка результатов работы в 4-м порядке теории возмущений (т. е. до членов $\sim a^5$) именно такой программы, где сумма по i в (7) (и соответственно цикл нахождения b_{ij}) начинается с 4. Если бы эти циклы начинались с 2, то результаты были бы те же, поскольку все b_{ij} с $i=2$ и 3 оказываются равными нулю, но сложность промежуточных выражений, а с ней и время счета, сильно возросли бы.

```
N:=4$
% NONLINEAR WATER WAVES ;
% PERTURBATION THEORY TO THE ORDER N ;
% FF—VELOCITY POTENTIAL, ZZ—WATER LEVEL ;
% BOUNDARY CONDITIONS ARE: ;
% SUB(Z=ZZ,DF(FF,T)+DF(FF,X)**2/2+DF(FF,Z)**2/2+Z)=0;
% DF(ZZ,T)=SUB(Z=ZZ,DF(FF,Z)-DF(FF,X)*DF(ZZ,X) ) ;
% WAVE VECTOR K=1, Y=X-V*T ;
SYMBOLIC OPERATOR TIME; TT:=TIME()$
OPERATOR B,EX; WEIGHT A=1; WTLEVEL N+1;
FACTOR A; ON DIV,RAT;

% GENERAL FORM OF THE SOLUTION FOR THE VELOCITY POTENCIAL ;
FF:=A*COS(Y)*E**Z+
  FOR I:=4:N+1 SUM A**I*
  IF FIXP(I/2)
```

```
THEN FOR J:=2 STEP 2 UNTIL I SUM B(I,J)*SIN(J*Y)*E**(J*Z)
ELSE FOR J:=3 STEP 2 UNTIL I SUM B(I,J)*COS(J*Y)*E**(J*Z)$
FZ:=DF(FF,Z)$
```

```
% TAYLOR EXPANSION ;
FOR ALL Z LET E**Z=EX(Z); F:=FF$ FZ:=FZ$
FOR ALL Z CLEAR E**Z;
FOR ALL K,Z LET EX(Z)**K=EX(K*Z); F:=F$ FZ:=FZ$
FOR ALL K,Z CLEAR EX(Z)**K;
FOR ALL K,Z LET A**K*EX(Z)=EX(N+1-K,Z); F:=F$ FZ:=FZ$
FOR ALL K,Z CLEAR A**K*EX(Z);
FOR ALL Z LET A*EX(Z)=EX(N,Z); F:=F$ FZ:=FZ$
FOR ALL Z CLEAR A*EX(Z);
FOR ALL K,Z LET EX(K,Z)=A**(N+1-K)*
BEGIN SCALAR A;A:=1;RETURN 1+FOR I:=1:K SUM A:=A*Z/I END;
F:=F$ FZ:=FZ$
FOR ALL K,Z CLEAR EX(K,Z);
LET DF(Y,X)=1,DF(Y,T)=-V; FX:=DF(F,X)$
FOR ALL X LET COS(X)**2=(1+COS(2*X))/2,
  SIN(X)**2=(1-COS(2*X))/2;
FOR ALL X,Y LET COS(X)*COS(Y)=(COS(X-Y)+COS(X+Y))/2,
  SIN(X)*SIN(Y)=(COS(X-Y)-COS(X+Y))/2,
  SIN(X)*COS(Y)=(SIN(X-Y)+SIN(X+Y))/2;

% WATER LEVEL FOUND FROM THE BOUNDARY CONDITION ;
WEIGHT Z=1; ZZ:=-DF(F,T)-FZ**2/2-FX**2/2$
REPEAT (<< CLEAR Z;ZZ1:=SUB(Z=Z1,ZZ);WEIGHT Z=1;
  ZZ:=SUB(Z1=ZZ,ZZ1) >>)
UNTIL FREEOF(ZZ,Z);

% THIS EXPRESSION SHOULD BE ZERO ;
% DUE TO THE SECOND BOUNDARY CONDITION ;
CLEAR Z;
Q:=DF(ZZ,T)-SUB(Z=ZZ,FZ)+SUB(Z=ZZ,FX)*DF(ZZ,X)$
CLEAR F,FZ,FX; CLEAR DF(Y,X),DF(Y,T);
FOR ALL X CLEAR COS(X)**2,SIN(X)**2;
FOR ALL X,Y CLEAR COS(X)*COS(Y),SIN(X)*SIN(Y),SIN(X)*COS(Y);
WEIGHT W=2; LET V**2=1+W; Q:=Q$ ZZ:=ZZ$ CLEAR V**2;
LET V=
BEGIN SCALAR A,B,J;A:=1;B:=3/2;J:=0;
RETURN 1+FOR I:=1:IF FIXP(N/2) THEN N/2 ELSE (N-1)/2 SUM
```

```

    << B:=B-1; J:=J+1; A:=A*B/J*W >>
END;
Q:=Q$ ZZ:=ZZ$ CLEAR V;

% RESONANT TERM ;
LET Z*COS(Y)=1; R:=Q*Z$ CLEAR Z*COS(Y);
R:=SUB(Z=0,R)$ Q:=Q-R*COS(Y)$
WW:=W-R/A$
REPEAT << CLEAR W; WW1:=SUB(W=W1,WW); WEIGHT W=2;
    WW:=SUB(W1=WW,WW1) >>
UNTIL FREEOF(WW,W);
CLEAR W; W:=WW$ CLEAR WW; Q:=Q$ ZZ:=ZZ$

FOR ALL J LET COS(J*Y)=Z**J, SIN(J*Y)=Z**J; Q:=Q$
FOR ALL J CLEAR COS(J*Y), SIN(J*Y);
ARRAY QA(1), QZ(1); CLEAR A;

PROCEDURE COEF1(Y,X,A);
BEGIN SCALAR N,D,M; N:=NUM(Y); D:=DEN(Y); CLEAR Y;
    M:=COEFF(N,X,A);
    FOR I:=0:M DO A(I):=A(I)/D; RETURN M
END;
COEF1

COEF1(Q,A,QA)$
FOR I:=4:N+1 DO
    << COEF1(QA(I),Z,QZ);
        FOR J:=I STEP -2 UNTIL 2 DO
            B(I,J):=-SUB(B(I,J)=0,QZ(J))/DF(QZ(J),B(I,J))
        >>;

% RESULTS ;
FF:=FF;
FF := -1/12*E(3*Z)*A5*COS(3*Y) - 1/2*E(2*Z)*A4*SIN(2*Y) + EZ*A*COS(Y)
ZZ:=ZZ;
ZZ := A5*(-125/384*SIN(5*Y) + 171/128*SIN(3*Y) - 121/192*SIN(Y)) +
A4*(1/3*COS(4*Y) - 5/6*COS(2*Y)) + A3*(3/8*SIN(3*Y) - 1/8*SIN(Y)) -
1/2*A2*COS(2*Y) - A*SIN(Y)
W:=1+W;

```

$$W := 3/2 * A^4 + A^2 + 1$$

```

WRITE "TIME ", TIME() - TT, " MSEC";
TIME 48457 MSEC
END;

```

Случай конечной глубины. В этом случае в общем виде потенциала скорости (7) вместо $\exp jz$ стоит $\operatorname{ch} j(h+z)$. Поэтому в программе описаны функции CH и SH и их необходимые свойства. Для запоминания того, до какого порядка нужно разлагать который экземпляр этих функций, в них вставляется дополнительный параметр, как и в предыдущей программе. В самих разложениях организуется необходимое чередование CH и SH.

Еще одно отличие связано с тем, что в линейном приближении $v^2 = thh \neq 1$. Поэтому мы теперь подставляем $V^{**2} = SH(H)/CH(H) * (1+W)$, где W — поправка 2 порядка малости. В линейном приближении резонансный член R теперь имеет вид $A * W * SH(H)$, поэтому изменена формула для WW. Здесь мы приводим результаты работы программы во 2-м порядке теории возмущений (т. е. до членов $\sim a^3$).

```

N:=2$
SYMBOLIC OPERATOR TIME; TT:=TIME()$
FACTOR A; ON DIV,RAT;
OPERATOR B,EX,CH,SH;
WEIGHT A=1; WTLEVEL N+1;
FOR ALL X LET DF(CH(X),X)=SH(X), DF(SH(X),X)=CH(X),
    SH(X)**2=CH(X)**2-1;

```

% GENERAL FORM OF THE SOLUTION FOR THE VELOCITY POTENTIAL ;

```

FF:=A*COS(Y)*CH(H+Z) +
FOR I:=2:N+1 SUM A**I*
IF FIXP(I/2)
THEN FOR J:=2 STEP 2 UNTIL I SUM
    B(I,J)*SIN(J*Y)*CH(J*(H+Z))
ELSE FOR J:=3 STEP 2 UNTIL I SUM
    B(I,J)*COS(J*Y)*CH(J*(H+Z))$
FZ:=DF(FF,Z)$

```

% TAYLOR EXPANSION ;

```

FOR ALL K,Z LET A**K*CH(Z)=CH(N+1-K,Z),
    A**K*SH(Z)=SH(N+1-K,Z);
FOR ALL Z LET A*CH(Z)=CH(N,Z), A*SH(Z)=SH(N,Z);

```

```

F:=FF$ FZ:=FZ$
FOR ALL K,Z CLEAR A**K*CH(Z),A**K*SH(Z);
FOR ALL Z CLEAR A*CH(Z),A*SH(Z);
FOR ALL K,J LET CH(K,J*H+J*Z) = A**(N+1-K)*
BEGIN SCALAR A,L;A:=1;L:=1;
  RETURN CH(J*H) + FOR I:=1:K SUM
  << A:=A*J*Z/I;L:=1-L;
  A*(IF L=0 THEN SH(J*H) ELSE CH(J*H)) >>
END;
FOR ALL K,J LET SH(K,J*H+J*Z) = A**(N+1-K)*
BEGIN SCALAR A,L;A:=1;L:=1;
  RETURN SH(J*H) + FOR I:=1:K SUM
  << A:=A*J*Z/I;L:=1-L;
  A*(IF L=0 THEN CH(J*H) ELSE SH(J*H)) >>
END;
FOR ALL K LET CH(K,H+Z) = A**(N+1-K)*
BEGIN SCALAR A,L;A:=1;L:=1;
  RETURN CH(H) + FOR I:=1:K SUM
  << A:=A*Z/I;L:=1-L;A*(IF L=0 THEN SH(H) ELSE CH(H)) >>
END;
FOR ALL K LET SH(K,H+Z) = A**(N+1-K)*
BEGIN SCALAR A,L;A:=1;L:=1;
  RETURN SH(H) + FOR I:=1:K SUM
  << A:=A*Z/I;L:=1-L;A*(IF L=0 THEN CH(H) ELSE SH(H)) >>
END;
F:=F$ FZ:=FZ$
FOR ALL K,J CLEAR CH(K,J*H+J*Z),SH(K,J*H+J*Z);
FOR ALL K CLEAR CH(K,H+Z),SH(K,H+Z);
FOR ALL N,X SUCH THAT FIXP(N) AND N>1 LET
  CH(N*X) = CH((N-1)*X)*CH(X) + SH((N-1)*X)*SH(X),
  SH(N*X) = CH((N-1)*X)*SH(X) + SH((N-1)*X)*CH(X);
F:=F$ FZ:=FZ$
FOR ALL N,X SUCH THAT FIXP(N) AND N>1 CLEAR CH(N*X),SH(N*X);
LET DF(Y,X) = 1,DF(Y,T) = -V; FX:=DF(F,X)$
FOR ALL X LET COS(X)**2 = (1+COS(2*X))/2,
  SIN(X)**2 = (1-COS(2*X))/2;
FOR ALL X,Y LET COS(X)*COS(Y) = (COS(X-Y) + COS(X+Y))/2,
  SIN(X)*SIN(Y) = (COS(X-Y) - COS(X+Y))/2,
  SIN(X)*COS(Y) = (SIN(X-Y) + SIN(X+Y))/2;

% WATER LEVEL FOUND FROM THE BOUNDARY CONDITION ;

```

```

WEIGHT Z=1; ZZ:= -DF(F,T) - FZ**2/2 - FX**2/2$
REPEAT << CLEAR Z;ZZ1:=SUB(Z=Z1,ZZ);WEIGHT Z=1;
  ZZ:=SUB(Z1=ZZ,ZZ1) >>
UNTIL FREEOF(ZZ,Z);

% THIS EXPRESSION SHOULD BE ZERO ;
% DUE TO THE SECOND BOUNDARY CONDITION ;
CLEAR Z;
Q:=DF(ZZ,T) - SUB(Z=ZZ,FZ) + SUB(Z=ZZ,FX)*DF(ZZ,X)$
CLEAR F,FZ,FX; CLEAR DF(Y,X),DF(Y,T);
FOR ALL X CLEAR COS(X)**2,SIN(X)**2;
FOR ALL X,Y CLEAR COS(X)*COS(Y),SIN(X)*SIN(Y),SIN(X)*COS(Y);
WEIGHT W=2; LET V**2=SH(H)/CH(H)*(1+W);
Q:=Q$ ZZ:=ZZ$ CLEAR V**2;
LET V=SQRT(SH(H)/CH(H))*
BEGIN SCALAR A,B,J;A:=1;B:=3/2;J:=0;
  RETURN 1 + FOR I:=1:IF FIXP(N/2) THEN N/2 ELSE (N-1)/2 SUM
  <<B:=B-1;J:=J+1;A:=A*B/J*W >>
END;
Q:=Q$ ZZ:=ZZ$ CLEAR V;

% RESONANT TERM ;
LET Z*COS(Y) = 1; R:=Q*Z$ CLEAR Z*COS(Y);
R:=SUB(Z=0,R)$ Q:=Q-R*COS(Y)$
WW:=W-R/(A*SH(H))$
REPEAT << CLEAR W;WW1:=SUB(W=W1,WW);WEIGHT W=2;
  WW:=SUB(W1=WW,WW1) >>
UNTIL FREEOF(WW,W);
CLEAR W; W:=WWS$ CLEAR WW; Q:=Q$ ZZ:=ZZ$

FOR ALL J LET COS(J*Y) = Z**J,SIN(J*Y) = Z**J;
Q:=Q$
FOR ALL J CLEAR COS(J*Y),SIN(J*Y);
ARRAY QA(1),QZ(1); CLEAR A;

PROCEDURE COEF1(Y,X,A);
BEGIN SCALAR N,D,M;N:=NUM(Y);D:=DEN(Y);CLEAR Y;
  M:=COEFF(N,X,A);
  FOR I:=0:M DO A(I):=A(I)/D;RETURN M
END;
COEF1

```



```

COEFL(Q,A,QA)$ ON GCD;
FOR I:=2:N+1 DO
  (<< COEFL(QA(I),Z,QZ);
    FOR J:=I STEP -2 UNTIL 2 DO
      B(I,J):=-SUB(B(I,J)=0,QZ(J))/DF(QZ(J),B(I,J))
    >>);
% RESULTS ;
FF:=FF;
FF:=A3*COS(3*Y)*SH(H)(-1)*CH(3*H+3*Z)*CH(H)*(1/16*CH(H)2-
13/64)/(CH(H)4-2*CH(H)2+1)+
(-3/8*SQRT(CH(H))(-1)*SQRT(SH(H))*A2*SIN(2*Y)*SH(H)(-1)*CH(2*H+
2*Z)*CH(H))/(CH(H)2-1)+A*COS(Y)*CH(H+Z)
ZZ:=ZZ;
ZZ:=SQRT(CH(H))(-1)*SQRT(SH(H))*A3*SH(H)(-1)*(3/8*SIN(3*Y)*CH(H)8+
3/64*SIN(3*Y)*CH(H)2-1/8*SIN(Y)*CH(H)8-1/4*SIN(Y)*CH(H)6+
9/16*SIN(Y)*CH(H)4-1/16*SIN(Y)*CH(H)2-1/8*SIN(Y))/(CH(H)4-
2*CH(H)2+1)+A2*(-1/2*COS(2*Y)*CH(H)4-1/4*COS(2*Y)*CH(H)2-
1/4*CH(H)2+1/4)/(CH(H)2-1)-
SQRT(CH(H))(-1)*SQRT(SH(H))*A*SIN(Y)*CH(H)
W:=1+W;
W:=A2*SH(H)*CH(H)(-1)*(CH(H)6-CH(H)4+7/8*CH(H)2
+1/4)/(CH(H)4-2*CH(H)2+1)+1
WRITE "TIME ",TIME()-TT," MSEC";
TIME 52690 MSEC
END;

```

4. ВЫЧИСЛЕНИЕ ТЕНЗОРА КРИВИЗНЫ

Введение. Наш очередной пример относится к общей теории относительности. Те, кто ее не знает, могут рассматривать его как упражнение по геометрии поверхностей в трехмерном пространстве. Математический аппарат в обоих случаях один и тот же — он

описывает искривленные пространства произвольной размерности. Более того, знание этого аппарата полезно и при работе в криволинейных координатах в обычном плоском пространстве. Он позволяет легко вывести формулы для дивергенции, ротора, лапласиана в цилиндрических, сферических и прочих координатах. Изложение этого математического аппарата можно найти в учебниках [4—6].

Мы будем использовать не совсем обычный язык — метод абстрактных индексов, принадлежащий Пенроузу [7]. Если усвоить этот метод, то все дальнейшие вычисления будут производиться автоматически и вероятность ошибок будет сведена к минимуму. Однако стиль изложения в этом параграфе может показаться излишне математизированным по сравнению с остальными. Тот, кто не чувствует себя достаточно подготовленным, может спокойно пропустить этот параграф.

Метод абстрактных индексов. Векторное пространство V над полем скаляров S (обычно это поле вещественных чисел R) — это такое множество, в котором определены операции сложения и умножения на скаляр, удовлетворяющие обычным свойствам.

Метод абстрактных индексов состоит в том, что мы вместо одного векторного пространства V будем рассматривать множество его копий V^a, V^b, V^c, \dots . Они совершенно одинаковы: каждому элементу $v^a \in V^a$ соответствует элемент $v^b \in V^b, v^c \in V^c, \dots$. Разумеется, сложение $u^a + v^a$ определено лишь для элементов одного и того же пространства V^a . Правильное векторное равенство также может связывать только элементы одного и того же пространства: $u^a = v^a$, или $u^b = v^b$, но не $u^a = v^b$. Абстрактные индексы a, b, c, \dots просто помечают одну из копий векторного пространства, и вовсе не пробегают каких либо числовых значений. Зачем вводить вместо одного пространства множество помеченных копий, выяснится чуть дальше, при рассмотрении тензоров.

Каждому пространству V^a соответствует сопряженное пространство V_a . Его элемент $u_a \in V_a$ есть линейное отображение $V^a \rightarrow S$, то есть он сопоставляет каждому вектору $v^a \in V^a$ скаляр $u_a v^a \in S$ (мы не пишем скобки в этом обозначении функции u_a от аргумента v^a). На множестве этих линейных отображений естественно вводится структура векторного пространства, если определить их сумму как $(u_a + w_a)v^a = u_a v^a + w_a v^a$ и умножение на скаляр как $(x u_a)v^a = x(u_a v^a)$. Каждый вектор $v^a \in V^a$ определяет линейное отображение $V_a \rightarrow S$ по формуле $v^a u_a = u_a v^a$, так что пространством, сопряженным к V_a , является V^a .

Тензоры. Теперь мы можем определить тензоры. Например,

тензор $t_a^b \in V_a^b$ — это отображение $V^a \times V_b \rightarrow S$, линейное по каждому аргументу. Оно сопоставляет каждой паре из вектора $v^a \in V^a$ и сопряженного вектора $u_b \in V_b$ скаляр $t_a^b v^a u_b \in S$, линейно зависящий как от v^a , так и от u_b . Выражение $t_a^b = p_a q^b$, где $p_a \in V_a$ и $q^b \in V^b$, является тензором; соответствующее отображение определяется как $(p_a q^b) v^a u_b = (p_a v^a) (q^b u_b)$. Тензор t_a^b можно также рассматривать как линейное отображение $V^a \rightarrow V^b$, поскольку при фиксированном v^a выражение $w^b = t_a^b v^a$ определяет линейное отображение $V_b \rightarrow S$, т. е. элемент $w^b \in V^b$. Аналогично, его можно считать и линейным отображением $V_b \rightarrow V_a$: $w_a = t_a^b v_b \in V_a$.

Точно так же определяются тензоры с любым числом индексов. Так, $t_{abcd} \in V_{abcd}^a$ определяется отображением $V_a \times V^b \times V^c \times V^d \rightarrow S$, линейным по каждому аргументу: $t_{abcd} u_a v^b w^c p^d$ является скаляром. Тензоры с одинаковым набором верхних и нижних индексов можно складывать. Любые тензоры, у которых все индексы различны, можно перемножать, получая тензоры с большим числом индексов: $t_{bcd}^a = s_{bc}^a r_{cd}$.

Кроме сложения и умножения, есть еще одна очень важная операция — свертка. Свертка производится по одному верхнему и одному нижнему индексу, и приводит к тензору без этих индексов. Например, свертка $t_{bac}^a = s_{bc} \in V_{bc}$. Повторяющийся индекс, обозначающий свертку, может быть любым, лишь бы он не совпадал ни с каким другим присутствующим в выражении индексом. Для тензора $t_a^b = p_a q^b$ свертка определяется как $t_a^a = p_a q^a$. Можно доказать, что любой тензор может быть представлен в виде

$$t_{abcd} = \sum_i (u_i)^a (v_i)^b (w_i)^c (p_i)^d$$

(одно из доказательств, использующее базис, будет приведено ниже). Тогда свертка определяется очевидно:

$$t_{bac}^a = \sum_i (u_i)^a (w_i)^d (v_i)^b (p_i)^c.$$

Теперь понятно, зачем нужно помечать индексами копии векторного пространства V . Если бы мы этого не делали, нам пришлось бы говорить что-нибудь вроде «свертка тензора t по третьему верхнему и пятому нижнему индексу», и мы бы очень скоро запутались. Обозначая индексы буквами, мы упрощаем их учет.

Важную роль играет единичный тензор $\delta_b^a \in V_b^a$. Как отображение $V_a \times V^b \rightarrow S$ он определяется равенством $\delta_b^a v_a u^b = v_a u^a = v_b u^b$. Он переводит $v^a \in V^a$ в соответствующий элемент другой копии

$v^b \in V^b$: $v^b = \delta_b^a v^a$. Точно так же, $v_b = \delta_b^a v_a$. Таким образом, он служит для замены индексов.

Базис и компоненты. До сих пор мы ни разу не упоминали базис, компоненты векторов и т. д. Теперь пора это сделать. Линейно независимый набор векторов $\delta_1^a, \delta_2^a, \dots, \delta_n^a \in V^a$ называется базисом, если любой вектор $v^a \in V^a$ линейно зависим от них, т. е. может быть по ним разложен: $v^a = v^1 \delta_1^a + v^2 \delta_2^a + \dots + v^n \delta_n^a$. Количество векторов n в любом базисе одинаково, и называется размерностью векторного пространства V .

Индекс, нумерующий векторы базиса и пробегающий значения 1, 2, ..., n , будем называть конкретным индексом и обозначать подчеркнутой буквой $\underline{a}, \underline{b}, \underline{c}, \dots$. Так, векторы базиса обобщенно обозначаются $\delta_{\underline{a}}^a$. По дважды повторяющемуся сверху и снизу конкретному индексу подразумевается суммирование от 1 до n , например, $v^a = v^{\underline{a}} \delta_{\underline{a}}^a$.

Базис $\delta_{\underline{a}}^a \in V^a$ однозначно определяет сопряженный базис $\delta_a^{\underline{a}} \in V_a$ в сопряженном пространстве. Он определяется условием $\delta_a^{\underline{a}} \delta_{\underline{a}}^a = \delta_{\underline{a}}^{\underline{a}}$, где

$$\delta_{\underline{a}}^{\underline{a}} = \begin{cases} 1, & \text{если } \underline{a} = \underline{b} \\ 0, & \text{если } \underline{a} \neq \underline{b} \end{cases}, \quad (1)$$

т. е. действие линейного отображения $\delta_{\underline{a}}^a$ на вектор v^a выделяет из него \underline{a} -ю компоненту: $v^{\underline{a}} = \delta_{\underline{a}}^a v^a$. Аналогично, $v_{\underline{a}} = \delta_a^{\underline{a}} v_a$. Условие полноты базиса, т. е. то, что любой вектор может быть по нему разложен, записывается как $v^a = v^{\underline{a}} \delta_{\underline{a}}^a = (\delta_b^{\underline{a}} v^b) \delta_{\underline{a}}^a$ для всех $v^a \in V^a$, то есть $\delta_{\underline{a}}^a \delta_b^{\underline{a}} = \delta_b^{\underline{a}}$.

Точно так же, как и вектор, любой тензор может быть разложен по базису: $t_{abcd}^a = t_{\underline{bcd}}^{\underline{a}} \delta_{\underline{a}}^a \delta_{\underline{b}}^b \delta_{\underline{c}}^c \delta_{\underline{d}}^d$. Напоминаем, что справа стоит сумма n^4 членов; компоненты $t_{\underline{bcd}}^{\underline{a}}$ — это n^4 чисел от t_{1111}^1 до t_{nnnn}^n . Компоненты выделяются действием на тензор базисных векторов: $t_{\underline{bcd}}^{\underline{a}} = t_{abcd}^a \delta_{\underline{a}}^a \delta_{\underline{b}}^b \delta_{\underline{c}}^c \delta_{\underline{d}}^d$.

Допустим, имеются 2 базиса $\delta_{\underline{a}}^a$ и $\delta_{\underline{a}}^{\underline{a}}$. Вектора нового базиса, очевидно, могут быть разложены по старому базису: $\delta_{\underline{a}}^{\underline{a}} = \delta_{\underline{a}}^a \delta_a^{\underline{a}}$, где $\delta_a^{\underline{a}} = \delta_{\underline{a}}^a \delta_a^{\underline{a}}$. Для сопряженного базиса $\delta_{\underline{a}}^a = \delta_{\underline{a}}^{\underline{a}} \delta_a^{\underline{a}}$, где $\delta_a^{\underline{a}} = \delta_{\underline{a}}^a \delta_a^{\underline{a}}$. Матрицы $\delta_{\underline{a}}^a$ и $\delta_a^{\underline{a}}$ являются взаимно обратными: $\delta_{\underline{a}}^a \delta_a^{\underline{a}} = \delta_{\underline{a}}^{\underline{a}}$, $\delta_a^{\underline{a}} \delta_{\underline{a}}^a = \delta_{\underline{a}}^{\underline{a}}$. Компоненты вектора $v^{\underline{a}}$ в новом базисе выражаются через компоненты в старом формулой $v^{\underline{a}} = v^a \delta_{\underline{a}}^a$; аналогично $v_{\underline{a}} = v_a \delta_a^{\underline{a}}$. Вообще, для любого тензора мы имеем, например, $t_{\underline{bcd}}^{\underline{a}} = t_{abcd}^a \delta_{\underline{a}}^a \delta_{\underline{b}}^b \delta_{\underline{c}}^c \delta_{\underline{d}}^d$. Таким образом, мы получили закон преобразования компонент тензора, с которого обычно начинается изучение тензорной алгебры.

Скалярное произведение. Важную роль играют векторные пространства, в которых определена операция скалярного произведения — отображение $V \times V \rightarrow S$, линейное по каждому аргументу. Если скалярное произведение симметрично относительно перестановки аргументов и положительно определено, то такое векторное пространство называется евклидовым; когда положительной определенности нет — псевдоевклидовым. Пример псевдоевклидова пространства — пространство-время Минковского в специальной теории относительности. Пространства с антисимметричным скалярным произведением называются симплектическими. К ним относится пространство спиноров, играющее важную роль в физике. Мы их рассматривать не будем.

Задание скалярного произведения означает, что задан тензор g_{ab} , называемый метрическим. Скалярное произведение векторов u^a и v^b равно $g_{ab}u^av^b$. Иначе можно сказать, что каждому вектору $u^a \in V^a$ естественным образом соответствует линейное отображение $V^b \rightarrow S$, задаваемое скалярным произведением $g_{ab}u^av^b$, т. е. вектор сопряженного пространства $u_b = g_{ab}u^a$. Это соответствие является взаимно однозначным. Обратное преобразование вектора из сопряженного пространства в исходное $u^a = g^{ab}u_b$ задается обратным тензором g^{ab} , определяемым условием $g^{ab}g_{bc} = \delta_c^a$. Тензор g^{ab} является метрическим (т. е. определяет скалярные произведения и длины) в сопряженном пространстве.

Метрические тензоры g_{ab} и g^{ab} позволяют опускать и поднимать индексы любого тензора: $t^a{}_{bcd} = g^{ae}t_{ebcd} = g_{ce}t^a{}_{b^e}{}_d = g_{be}g_d{}^f t^{ae}{}_c{}^f \dots$. Все эти тензоры содержат одинаковую информацию и обозначаются одной буквой. Именно по этой причине мы следим за порядком индексов, даже когда одни из них верхние, а другие нижние: все их можно опустить или поднять. Исключением из этого правила являются симметричные тензоры: если $t_{ab} = t_{ba}$, то все равно, поднимать первый или второй индекс, и можно писать t_b^a . Если поднять индекс в g_{ab} или опустить в g^{ab} , получится δ_b^a .

Многообразия. Бесконечно малые участки n -мерного многообразия устроены так же, как у плоского n -мерного пространства. Примером 2-мерных многообразий являются поверхности в 3-мерном пространстве.

Если рассмотреть точку x многообразия, то из нее в бесконечно близкие точки можно провести векторы dx^a , лежащие в касательном пространстве $T^a(x)$. Векторы касательного пространства, не являющиеся бесконечно малыми, можно получить, например, как касательные к кривым. Если $x(t)$ — кривая (t — параметр), то

касательный вектор $dx^a/dt|_{t=t_0} \in T^a(x(t_0))$. В случае 2-мерной поверхности наглядно видно, что касательные векторы ко всевозможным кривым в точке x образуют касательную плоскость.

Градиент скалярной функции $\nabla_a \phi$ принадлежит сопряженному касательному пространству $T_a(x)$, поскольку $d\phi = \nabla_a \phi dx^a$ — скаляр.

Вообще, на многообразии можно рассматривать любые тензорные поля, например $t^a{}_{bcd}(x)$. Значение этого поля в точке x_0 принадлежит соответствующему тензорному произведению касательного пространства и сопряженного к нему: $t^a{}_{bcd}(x_0) \in T^a{}_{bcd}(x_0)$.

Координаты. В некоторой области n -мерного многообразия можно ввести координаты — n гладких функций x^a , однозначно определяющих точку. Не всегда удается ввести единую систему координат на всем многообразии, например, на сфере это невозможно. Подчеркнем, что x^a не являются компонентами вектора.

После введения координат можно определить естественный базис $\delta_a^a = \nabla_a x^a$ в $T_a(x)$, и соответствующий базис δ_a^a в $T^a(x)$, определяемый условием $\delta_a^a \delta_b^a = \delta_b^a$. В этом базисе $\nabla_a x^b = \delta_a^b$, поэтому оператор градиента имеет вид $\nabla_a = \partial/\partial x^a$.

Иногда, введя систему координат x^a , используют другие, неестественные базисы в касательном пространстве. Так, в криволинейных координатах в 3-мерном евклидовом пространстве вместо $\delta_a^a = \nabla_a x^a$ обычно используют векторы единичной длины. В таких случаях дальнейшие формулы потребуют некоторого несложного обобщения.

Если вместо старой системы координат x^a ввести новую $\tilde{x}^a(x^a)$, то компоненты тензоров преобразуются от старого естественного базиса к новому как

$$t^{\tilde{a}}{}_{\tilde{b}\tilde{c}\tilde{d}} = t^a{}_{bcd} \delta_{\tilde{a}}^a \delta_{\tilde{b}}^b \delta_{\tilde{c}}^c \delta_{\tilde{d}}^d,$$

где $\delta_{\tilde{a}}^a = \nabla_a \tilde{x}^{\tilde{a}}$, $\delta_{\tilde{a}}^a = \nabla_{\tilde{a}} x^a$. Эти матрицы взаимно обратны: $\delta_{\tilde{a}}^a \delta_{\tilde{a}}^{\tilde{b}} = \delta_{\tilde{a}}^{\tilde{b}}$, $\delta_{\tilde{a}}^a \delta_{\tilde{b}}^a = \delta_{\tilde{b}}^a$.

Связность. До сих пор мы умели дифференцировать только скаляры, но не векторы и тензоры. Векторы, определенные в соседних точках, лежат в разных, не связанных друг с другом касательных пространствах, и их нельзя вычитать.

Можно ввести на многообразии дополнительную структуру, называемую связностью — правило дифференцирования векторных полей. Разумеется, на данном многообразии можно ввести много разных связностей.

Как найти компоненты $\nabla_b v^a$ в заданной системе координат? Поскольку любой вектор разлагается по базисным: $v^a = v^a \delta_a^a$, достаточно уметь дифференцировать базисные векторы, т. е. знать компоненты тензора

$$\nabla_b \delta_a^a = \Gamma_{ab}^a. \quad (2)$$

Этот тензор различен в разных системах координат. Его компоненты Γ_{bc}^a называются коэффициентами связности (или символом Кристоффеля). Зная их, можно дифференцировать любой вектор:

$$\begin{aligned} \nabla_b v^a &= \nabla_b (v^a \delta_a^a) = \delta_a^a \nabla_b v^a + \Gamma_{ab}^a v^a, \\ \delta_a^a \delta_b^b \nabla_b v^a &= \nabla_b v^a + \Gamma_{ab}^a v^a. \end{aligned} \quad (3)$$

Чтобы продифференцировать вектор $v_a(x)$, нужно научиться дифференцировать базисные векторы δ_a^a . Они определяются соотношением $\delta_a^a \delta_{a'}^a = \delta_{a'}^a$. Применяв к нему ∇_b , получим $\delta_a^a \Gamma_{a'b}^a + \delta_{a'}^a \nabla_b \delta_a^a = 0$ (так как $\delta_{a'}^a$ — константы), или

$$\nabla_b \delta_a^a = -\Gamma_{ab}^a. \quad (4)$$

Теперь мы можем найти компоненты $\nabla_b v_a$:

$$\begin{aligned} \nabla_b v_a &= \nabla_b (v_a \delta_a^a) = \delta_a^a \nabla_b v_a - \Gamma_{ab}^a v_a, \\ \delta_b^b \delta_a^a \nabla_b v_a &= \nabla_b v_a - \Gamma_{ab}^a v_a. \end{aligned} \quad (5)$$

Точно так же находятся компоненты производной любого тензора, например $\nabla_d t^{ab}_c$:

$$\begin{aligned} \nabla_d t^{ab}_c &= \nabla_d (t^{ab}_c \delta_a^a \delta_b^b \delta_c^c) = \delta_a^a \delta_b^b \delta_c^c \nabla_d t^{ab}_c + \\ &+ t^{ab}_c (\Gamma_{ad}^a \delta_b^b \delta_c^c + \delta_a^a \Gamma_{bd}^b \delta_c^c - \delta_a^a \delta_b^b \Gamma_{cd}^c), \\ \delta_a^a \delta_b^b \delta_c^c \nabla_d t^{ab}_c &= \nabla_d t^{ab}_c + \Gamma_{ad}^a t^{ab}_c + \Gamma_{bd}^b t^{ab}_c - \Gamma_{cd}^c t^{ab}_c. \end{aligned} \quad (6)$$

Напомним, что оператор ∇_a , действующий на скаляр, т. е. на выражение без абстрактных индексов, есть просто $\partial/\partial x^a$.

Кривизна. Рассмотрим выражение $(\nabla_a \nabla_b - \nabla_b \nabla_a) \varphi$. В нем члены со вторыми производными сокращаются, и оно линейно по $\nabla \varphi$, т. е. имеет вид $K^c_{ab} \nabla_c \varphi$. Тензор K^c_{ab} называется кривизной, он антисимметричен по a и b . Найдем его компоненты:

$$\begin{aligned} \delta_a^a \delta_b^b (\nabla_a \nabla_b - \nabla_b \nabla_a) \varphi &= \nabla_a \nabla_b \varphi - \Gamma_{ba}^c \nabla_c \varphi - (a \leftrightarrow b) = \\ &= (\Gamma_{ab}^c - \Gamma_{ba}^c) \nabla_c \varphi, \quad K^c_{ab} = \Gamma_{ab}^c - \Gamma_{ba}^c. \end{aligned} \quad (7)$$

В дальнейшем мы будем рассматривать только связности без кручения. У них коэффициенты Γ_{bc}^a симметричны по b и c . При наличии кручения дальнейшие формулы требуют некоторого обобщения.

Рассмотрим теперь выражение $(\nabla_a \nabla_b - \nabla_b \nabla_a) v^c$. В нем вторые производные сокращаются; в силу отсутствия кручения, сокращаются и первые, и остается выражение, линейное по v :

$$(\nabla_a \nabla_b - \nabla_b \nabla_a) v^c = R^c_{cab} v^c, \quad (8)$$

где тензор R^c_{cab} называется кривизной (или тензором Римана). Он антисимметричен по a и b .

Через него же выражается $(\nabla_a \nabla_b - \nabla_b \nabla_a) v_c$. Действительно, рассмотрим выражение $(\nabla_a \nabla_b - \nabla_b \nabla_a) (v_c u^c)$, равное нулю. В нем члены с первыми производными v и u сокращаются, и остается $v_c (\nabla_a \nabla_b - \nabla_b \nabla_a) u^c + u^c (\nabla_a \nabla_b - \nabla_b \nabla_a) v_c$. Первое слагаемое равно $v_c R^c_{cab} u^c = u^c R^c_{cab} v_c$, откуда

$$(\nabla_a \nabla_b - \nabla_b \nabla_a) v_c = -R^c_{cab} v_c. \quad (9)$$

Точно так же, для любого тензора мы имеем, например,

$$(\nabla_a \nabla_b - \nabla_b \nabla_a) t^{cd}_e = R^c_{cab} t^{cd}_e + R^d_{dab} t^{cd}_e - R^e_{cab} t^{cd}_e. \quad (10)$$

Если тензор кривизны равен нулю, то многообразие является плоским в смысле его внутренней геометрии (например, цилиндр, который может быть получен свертыванием плоского листа бумаги). Из тензора Римана сверткой получается тензор Риччи $R_{ab} = R^c_{acb}$, содержащий меньше информации (в частности, может быть $R_{ab} = 0$, но $R^a_{bcd} \equiv 0$).

Найдем выражение для тензора кривизны, подставив в (8) разложение $v^c = v^d \delta_d^c$. Первые производные v^d и δ_d^c сократятся; вторые производные v^d дадут 0 в силу отсутствия кручения, и останется

$$R^c_{dab} = (\nabla_a \nabla_b - \nabla_b \nabla_a) \delta_d^c. \quad (11)$$

Продолжим вычисление:

$$\begin{aligned} R^c_{dab} &= \nabla_a \Gamma^c_{db} - (a \leftrightarrow b) = \nabla_a (\Gamma^c_{db} \delta_c^c \delta_b^b) - (a \leftrightarrow b) = \\ &= \delta_c^c \delta_b^b \nabla_a \Gamma^c_{db} + \Gamma^c_{db} (\Gamma^c_{ca} \delta_b^b - \delta_c^c \Gamma^b_{ba}) - (a \leftrightarrow b). \end{aligned}$$

Окончательно

$$R^c_{dad} = \nabla_a \Gamma^c_{db} - \nabla_b \Gamma^c_{da} + \Gamma^c_{ea} \Gamma^e_{db} - \Gamma^c_{eb} \Gamma^e_{da}. \quad (12)$$

Из этой явной формулы можно получить еще одно свойство симметрии: $R^a_{bcd} + R^a_{cdb} + R^a_{abc} = 0$.

Метрика. Если на многообразии задано поле метрического тензора $g_{ab}(x)$, так что касательные пространства являются (псевдо-) евклидовыми, то само многообразие называется (псевдо-) римановым пространством. В нем определен квадрат длины любого бесконечно малого отрезка dx^a :

$$ds^2 = g_{ab} dx^a dx^b. \quad (13)$$

Пространство-время в общей теории относительности является псевдоримановым.

В (псевдо-) римановом пространстве существует единственная связность без кручения, согласованная с метрикой в том смысле, что $\nabla_a g_{bc} = 0$. Записывая это условие в компонентах, мы получим

$$\nabla_a g_{bc} - \Gamma^{b'}_{ba} g_{b'c} - \Gamma^{c'}_{ca} g_{bc'} = 0. \quad (14)$$

Отсюда следует уравнение для Γ со всеми нижними индексами. Запишем его, а также еще 2 уравнения, получающиеся при переименовании индексов, и получим систему

$$\begin{aligned} \Gamma_{abc} + \Gamma_{bac} &= \nabla_c g_{ab}, \\ \Gamma_{acb} + \Gamma_{cab} &= \nabla_b g_{ac}, \\ \Gamma_{bca} + \Gamma_{cba} &= \nabla_a g_{bc}. \end{aligned} \quad (15)$$

Если, учитывая симметрию $\Gamma_{abc} = \Gamma_{acb}$, сложить 2 первых уравнения и вычесть третье, то мы получим окончательно

$$\Gamma_{abc} = \frac{1}{2} (\nabla_b g_{ac} + \nabla_c g_{ab} - \nabla_a g_{bc}). \quad (16)$$

При наличии метрики тензор кривизны имеет дополнительные свойства симметрии. Именно, из тождества $(\nabla_a \nabla_b - \nabla_b \nabla_a) g_{cd} = 0$ с учетом (9) следует $-R^c_{cab} g^{cd} - R^d_{dab} g_{cd} = 0$, или $R_{dcab} = -R_{cdab}$. С учетом ранее полученных свойств, отсюда следует также, что $R_{cdab} = R_{abcd}$. Поэтому тензор Риччи симметричен: $R_{ab} = R_{ba}$. Можно ввести скалярную кривизну $R = R^a_a$. В общей теории относительности важную роль играет тензор Эйнштейна $G_{ab} = R_{ab} - Rg_{ab}/2$. Уравнение Эйнштейна утверждает, что он равен (с точностью до

постоянного множителя) тензору энергии-импульса материи, создающей гравитационное поле.

Алгоритм. Теперь понятен алгоритм вычисления тензора кривизны по метрике. Пусть в некоторой системе координат заданы компоненты метрического тензора g_{ab} . Дальше действуем так:

1. Находим g^{ab} как обратную матрицу к g_{ab} .
2. Вычисляем Γ_{abc} по формуле (16).
3. Получаем $\Gamma^a_{bc} = g^{ad} \Gamma_{dbc}$.
4. Вычисляем R_{abcd} по формуле (12) с опущенным первым индексом:

$$R_{abcd} = g_{ae} (\nabla_c \Gamma^e_{bd} - \nabla_d \Gamma^e_{bc}) + \Gamma_{aec} \Gamma^e_{bd} - \Gamma_{aed} \Gamma^e_{bc}.$$

5. Находим $R_{ab} = g^{cd} R_{cabd}$.
6. Находим $R = g^{ab} R_{ab}$.
7. Получаем $G_{ab} = R_{ab} - Rg_{ab}/2$.

Программа. Здесь воспроизводится с небольшими изменениями программа из стандартного теста системы REDUCE, написанная А. Херном, Д. Бартоном и Дж. Фитчем.

Чтобы не делать лишней работы, нужно вычислять не все компоненты, а только независимые. Поскольку $\Gamma_{kij} = \Gamma_{kji}$, достаточно вычислять Γ_{kij} с $j \geq i$, остальные компоненты получаются заменой $i \leftrightarrow j$.

Тензор кривизны обладает сложными свойствами симметрии: $R_{ijkl} = -R_{jikl} = -R_{ijlk} = R_{klij}$. В силу антисимметрии по первой и второй парам индексов, достаточно вычислять компоненты с $j > i$ и $l > k$. В силу симметрии относительно перестановки пар индексов, при этом можно ограничиться компонентами с $k \geq i$; более того, при $k = i$ достаточно членов с $l \geq j$. Именно так и устроены циклы в программе. Вычислив компоненту, которая удовлетворяет этим условиям, мы получаем из нее еще 3 при помощи антисимметрии по парам индексов. Если перестановка пар дает новые компоненты (т. е. при $i \neq k$ или $j \neq l$), то нужно запомнить и еще 4 компоненты.

При вычислении скалярной кривизны свертываются 2 симметричных тензора g^{ij} и R_{ij} . При $i \neq j$ в сумме есть 2 одинаковых члена, отличающихся заменой $i \leftrightarrow j$. Чтобы не вычислять их дважды, сумма берется по $j \geq i$, а члены с $j \neq i$ удваиваются.

В простых случаях очень многие компоненты Γ_{kij} , R_{ijkl} и т. д. равны нулю. Чтобы не смотреть на многочисленные присваивания вида $\dots := 0$, включен флаг NERO (No zERO), подавляю-

щий их вывод. Кроме того, чтобы не вычислять производные Γ , которые умножатся на нулевые компоненты g_{lm} , используется условное выражение.

В качестве примера рассмотрена метрика Шварцшильда, описывающая гравитационное поле сферически симметричного тела вне его (в частности, поле черной дыры):

$$ds^2 = \left(1 - \frac{r_0}{r}\right) dt^2 - \frac{dr^2}{1 - \frac{r_0}{r}} - r^2(d\theta^2 + \sin^2\theta d\phi^2). \quad (17)$$

Здесь r_0 — гравитационный радиус. В программе он взят за 1 (т. е. r измеряется в единицах r_0). Эта метрика удовлетворяет уравнению Эйнштейна в вакууме, поэтому для нее $R_{ij} = 0$, $R = 0$, $G_{ij} = 0$.

```
% COMPUTATION OF THE CURVATURE TENSOR ;
% THE PROGRAM FROM THE STANDARD REDUCE TEST ;
% WRITTEN BY ANTHONY HEARN, DAVID BARTON AND JOHN FITCH ;
% WITH SMALL MODIFICATIONS ;
SYMBOLIC OPERATOR TIME; TT:=TIME()$
% COORDINATES ;
OPERATOR X;
% THE METRIC TENSOR ;
N:=3$ ARRAY G1(N,N);
% THE METRIC TENSOR SHOULD BE GIVEN HERE ;
G1(0,0):=1-1/X(1)$
G1(1,1):=-1/(1-1/X(1))$
G1(2,2):=-X(1)**2$
G1(3,3):=-X(1)**2*SIN(X(2))**2$
```

```
% THIS FLAG SUPPRESSES THE PRINTOUT OF ASSIGNMENTS ... := 0;
ON NERO; ON GCD;
```

```
% COMPUTATION OF THE METRIC TENSOR WITH UPPER INDICES G2 ;
ARRAY G2(N,N); MATRIX M1(N+1,N+1),M2(N+1,N+1);
FOR I:=0:N DO FOR J:=0:N DO M1(I+1,J+1):=G1(I,J);
M2:=1/M1$
FOR I:=0:N DO FOR J:=0:N DO WRITE G2(I,J):=M2(I+1,J+1);
G2(0,0) := X(1)/(X(1)-1)
G2(1,1) := (-X(1)+1)/X(1)
G2(2,2) := (-1)/X(1)**2
```

```
G2(3,3) := (-1)/(X(1)**2*SIN(X(2))**2)
CLEAR M1,M2;
% COMPUTATION OF THE CHRISTOFFEL SYMBOLS C1 AND C2 ;
ARRAY C1(N,N,N),C2(N,N,N);
FOR I:=0:N DO FOR J:=I:N DO
<< FOR K:=0:N DO WRITE C1(K,I,J):=C1(K,J,I):=
(DF(G1(I,K),X(J))+DF(G1(J,K),X(I))-DF(G1(I,J),X(K)))/2;
FOR K:=0:N DO WRITE C2(K,I,J):=C2(K,J,I):=
FOR L:=0:N SUM G2(K,L)*C1(L,I,J)
>>);
C1(1,0,0) := C1(1,0,0) := (-1)/(2*X(1)**2)
C2(1,0,0) := C2(1,0,0) := (X(1)-1)/(2*X(1)**3)
C1(0,0,1) := C1(0,1,0) := 1/(2*X(1)**2)
C2(0,0,1) := C2(0,1,0) := 1/(2*X(1)*(X(1)-1))
C1(1,1,1) := C1(1,1,1) := 1/(2*(X(1)**2-2*X(1)+1))
C2(1,1,1) := C2(1,1,1) := (-1)/(2*X(1)*(X(1)-1))
C1(2,1,2) := C1(2,2,1) := -X(1)
C2(2,1,2) := C2(2,2,1) := 1/X(1)
C1(3,1,3) := C1(3,3,1) := -X(1)*SIN(X(2))**2
C2(3,1,3) := C2(3,3,1) := 1/X(1)
C1(1,2,2) := C1(1,2,2) := X(1)
C2(1,2,2) := C2(1,2,2) := -X(1)+1
C1(3,2,3) := C1(3,3,2) := -X(1)**2*COS(X(2))*SIN(X(2))
C2(3,2,3) := C2(3,3,2) := COS(X(2))/SIN(X(2))
C1(1,3,3) := C1(1,3,3) := X(1)*SIN(X(2))**2
C1(2,3,3) := C1(2,3,3) := X(1)**2*COS(X(2))*SIN(X(2))
C2(1,3,3) := C2(1,3,3) := SIN(X(2))**2*(-X(1)+1)
C2(2,3,3) := C2(2,3,3) := -COS(X(2))*SIN(X(2))
% COMPUTATION OF THE RIEMANN TENSOR ;
ARRAY R(N,N,N,N);
FOR I:=0:N DO FOR J:=I+1:N DO FOR K:=I:N DO
FOR L:=K+1:N IF K=I THEN J ELSE N DO
<< WRITE R(I,J,K,L):=R(J,I,L,K):=
FOR M:=0:N SUM IF G1(I,M)=0 THEN 0 ELSE
G1(I,M)*(DF(C2(M,J,L),X(K))-DF(C2(M,J,K),X(L)))
```

```

+FOR M:=0:N SUM (C1(I,M,K)*C2(M,J,L) - C1(I,M,L)*C2(M,J,K));
R(I,J,L,K):=R(J,I,K,L):=-R(I,J,K,L);
IF I NEQ K OR J > L THEN
<< R(K,L,I,J):=R(L,K,J,I):=R(I,J,K,L);
R(L,K,I,J):=R(K,L,J,I):=-R(I,J,K,L)
>>
>>
R(0,1,0,1) := R(1,0,1,0) := 1/X(1)3
R(0,2,0,2) := R(2,0,2,0) := (-X(1)+1)/(2*X(1)2)
R(0,3,0,3) := R(3,0,3,0) := (SIN(X(2))2*(-X(1)+1))/(2*X(1)2)
R(1,2,1,2) := R(2,1,2,1) := 1/(2*(X(1)-1))
R(1,3,1,3) := R(3,1,3,1) := SIN(X(2))2/(2*(X(1)-1))
R(2,3,2,3) := R(3,2,3,2) := -X(1)*SIN(X(2))2
% COMPUTATION OF THE RICCI TENSOR
ARRAY RICCI(N,N);
FOR I:=0:N DO
WRITE RICCI(I,J):=RICCI(J,I)
FOR K:=0:N SUM FOR L:=0:N SUM G2(K,L)*R(K,I,L,J);

% COMPUTATION OF THE RICCI SCALAR ;
RS:=FOR I:= 0:N SUM FOR J:= I:N SUM
(IF J=I THEN 1 ELSE 2)*G2(I,J)*RICCI(I,J);

% COMPUTATION OF THE EINSTEIN TENSOR ;
ARRAY EINSTEIN(N,N);
FOR I:=0:N DO FOR J:=I:N DO
WRITE EINSTEIN(I,J):=EINSTEIN(J,I):=
RICCI(I,J) - RS*G1(I,J)/2;
% END OF THE PROGRAM ;
CLEAR G1,G2,C1,C2,R,RICCI,RS,EINSTEIN;
WRITE "TIME ",TIME() - TT, "MSEC";
TIME 27400 MSEC

```

Задача 1. Рассмотрите метрику замкнутого мира Фридмана:

$$ds^2 = dt^2 - a^2(t)(d\chi^2 + \sin^2\chi(d\theta^2 + \sin^2\theta d\varphi^2)). \quad (18)$$

Здесь $a(t)$ — радиус Вселенной. Приравняв найденный в программе тензор Эйнштейна тензору энергии-импульса материи, можно получить уравнение для $a(t)$.

Задача 2. Докажите, что тензор кривизны плоского 3-мерного пространства в цилиндрических ($ds^2 = dz^2 + dr^2 + r^2 d\varphi^2$) и в сферических ($ds^2 = dr^2 + r^2(d\theta^2 + \sin^2\theta d\varphi^2)$) координатах равен нулю. Найдите тензор кривизны сферы ($ds^2 = a^2(d\theta^2 + \sin^2\theta d\varphi^2)$), a — радиус сферы; его можно принять за единицу).

5. СЛОЖЕНИЕ МОМЕНТОВ В КВАНТОВОЙ МЕХАНИКЕ

Введение. В этом параграфе мы рассмотрим задачу сложения моментов в квантовой механике. Свойства оператора момента описаны в любом учебнике квантовой механики, например, в §§ 26—27 в [8] или в главе 4 в [9]. Имеется очень простой алгоритм сложения моментов, основанный на повышающих и понижающих операторах, см., например, [9]. В действительности существует решение в виде явной формулы, приведенное в § 106 учебника [8]. Однако пользоваться этой формулой довольно неудобно, так как она содержит сложные суммы. Обычно удобнее на практике применять таблицы, которые публикуются, например, в книжечках «REVIEW OF PARTICLE PROPERTIES», выходящих каждые 2 года. Еще удобнее иметь под рукой действующую программу, которую можно использовать в любой момент. В этом параграфе мы и приводим такую программу вместе с таблицей сложения моментов вплоть до 2, полученной с ее помощью.

Момент в квантовой механике. Оператор момента \hat{J} определяется так, что $\hat{U} = \exp(-i\hat{J}\delta\vec{\varphi})$ является оператором бесконечно малого поворота на $\delta\vec{\varphi}$ (вектор $\delta\vec{\varphi}$ равен по величине углу поворота, и направлен по его оси, согласно правилу буравчика). Т. е. если $|\psi\rangle$ — некоторое состояние, то $\hat{U}|\psi\rangle$ есть то же состояние, повернутое на $\delta\vec{\varphi}$. Поэтому среднее \bar{V}' от любого векторного оператора \hat{V} по $\hat{U}|\psi\rangle$ связано со средним \bar{V} от того же оператора по $|\psi\rangle$ формулой $\bar{V}' = \bar{V} + \delta\vec{\varphi} \times \bar{V}$. Поэтому $\hat{U} + \hat{V}\hat{U} = \hat{V} + i[\hat{J}\delta\vec{\varphi}, \hat{V}] = \hat{V} + \delta\vec{\varphi} \times \hat{V}$. Значит, для любого векторного оператора справедливо коммутационное соотношение $[\hat{V}_i, \hat{J}_j] = i\epsilon_{ijk}\hat{V}_k$. В частности, сам момент является векторным оператором; поэтому

$$[\hat{J}_i, \hat{J}_j] = i\epsilon_{ijk}\hat{J}_k. \quad (1)$$

Отсюда следует $[\hat{J}^2, \hat{J}_i] = 0$, где $\hat{J}^2 = \hat{J}_x^2 + \hat{J}_y^2 + \hat{J}_z^2$. Значит, существует система общих собственных состояний операторов \hat{J}^2 и

\hat{f}_z . Введем операторы $\hat{f}_{\pm} = \hat{f}_x \pm i\hat{f}_y$, $\hat{f}_{\pm}^{\pm} = \hat{f}_{\mp}$; для них из (1) получим $[\hat{f}_z, \hat{f}_{\pm}] = \pm \hat{f}_{\pm}$. Это значит, что если $|\psi\rangle$ — собственное состояние \hat{f}_z : $\hat{f}_z|\psi\rangle = m|\psi\rangle$, то $\hat{f}_{\pm}|\psi\rangle$ — тоже собственные состояния \hat{f}_z : $\hat{f}_z\hat{f}_{\pm}|\psi\rangle = (m \pm 1)\hat{f}_{\pm}|\psi\rangle$, (если только они не равны нулю). Значит, собственные значения оператора \hat{f}_z образуют последовательность с шагом 1, а \hat{f}_{\pm} являются повышающим и понижающим оператором.

Рассмотрим состояния с данным собственным значением \hat{f}_z . Для них собственные значения \hat{f}_z ограничены сверху и снизу, так как оператор $\hat{f}^2 - \hat{f}_z^2 = \hat{f}_x^2 + \hat{f}_y^2$ положительно определен. Пусть $|m_{\pm}\rangle$ — состояния с максимальным и минимальным собственными значениями \hat{f}_z , равными m_{\pm} . Тогда эти собственные значения нельзя соответственно повысить и понизить операторами \hat{f}_{\pm} : $\hat{f}_{\pm}|m_{\pm}\rangle = 0$. С учетом (1) мы имеем

$$\hat{f}_{\pm}\hat{f}_{\mp} = \hat{f}^2 - \hat{f}_z^2 \pm \hat{f}_z. \quad (2)$$

Отсюда $\hat{f}_{\mp}\hat{f}_{\pm}|m_{\pm}\rangle = 0 = [\hat{f}^2 - m_{\pm}(m_{\pm} \pm 1)]|m_{\pm}\rangle$, т. е. собственное значение оператора \hat{f}^2 для этих (и остальных рассматриваемых) состояний равно $m_{+}(m_{+} + 1) = m_{-}(m_{-} - 1)$. Значит, $(m_{+} + m_{-})(m_{+} - m_{-} + 1) = 0$, откуда с учетом $m_{+} \geq m_{-}$ находим $m_{-} = -m_{+}$, или $m_{\pm} = \pm j$. Поскольку m_{+} и m_{-} должны отличаться на целое число, j должно быть целым или полуцелым.

Итак, мы имеем общую систему собственных состояний $|j, m\rangle$ операторов \hat{f}^2 и \hat{f}_z :

$$\begin{aligned} \hat{f}^2|j, m\rangle &= j(j+1)|j, m\rangle, \\ \hat{f}_z|j, m\rangle &= m|j, m\rangle, \end{aligned} \quad (3)$$

где j — целое или полуцелое, а m меняется от $-j$ до j с шагом 1. Операторы \hat{f}_{\pm} повышают и понижают m : $\hat{f}_{\pm}|j, m\rangle = a_{\pm}(j, m)|j, m \pm 1\rangle$. Подбирая фазы состояний $|j, m\rangle$, можно сделать $a_{\pm}(j, m)$ действительными и положительными. Они находятся из условия нормировки: $|a_{\pm}(j, m)|^2 = \langle j, m | \hat{f}_{\mp} \hat{f}_{\pm} | j, m \rangle = j(j+1) - m(m \pm 1)$. Поэтому окончательно мы получаем

$$\begin{aligned} \hat{f}_{\pm}|j, m\rangle &= \sqrt{j(j+1) - m(m \pm 1)}|j, m \pm 1\rangle \\ &= \sqrt{(j \pm m + 1)(j \mp 1)}|j, m \pm 1\rangle. \end{aligned} \quad (4)$$

Сложение моментов. Пусть теперь мы имеем 2 оператора

момента \hat{f}_1 и \hat{f}_2 , коммутирующих друг с другом. Тогда существует базис $|j_1, m_1; j_2, m_2\rangle$ общих собственных состояний операторов $\hat{f}_1^2, \hat{f}_{1z}, \hat{f}_2^2, \hat{f}_{2z}$. С другой стороны, оператор полного момента $\hat{f} = \hat{f}_1 + \hat{f}_2$ также является моментом. Поэтому из состояний $|j_1, m_1; j_2, m_2\rangle$ при данных j_1, j_2 можно построить линейные комбинации $|j, m\rangle$, являющиеся собственными состояниями \hat{f}^2 и \hat{f}_z . Эта задача называется сложением моментов j_1 и j_2 .

Поскольку $\hat{f}_z = \hat{f}_{1z} + \hat{f}_{2z}$, всегда $m = m_1 + m_2$. Как видно из рис. 2 (где для определенности считается $j_1 \geq j_2$), имеется одно состояние с $m = j_1 + j_2$, два состояния с $m = j_1 + j_2 - 1$ и т. д. Такое




Рис. 2.

увеличение числа состояний происходит вплоть до $m = j_1 - j_2$, далее оно остается постоянным до $m = -(j_1 - j_2)$, и затем убывает до 1 при $m = -(j_1 + j_2)$. Значит, максимальный момент, который можно получить при сложении j_1 и j_2 , равен $j = j_1 + j_2$. Одно состояние в двумерном пространстве состояний с $m = j_1 + j_2 - 1$ относится к этому же моменту, а другое является состоянием с

максимальной проекцией для момента $j = j_1 + j_2 - 1$. Продолжая такие рассуждения, мы увидим, что возникают моменты вплоть до $j_1 - j_2$. Итак, в общем случае при сложении моментов j_1 и j_2 получаются моменты j от $j_1 + j_2$ до $|j_1 - j_2|$ через 1.

Алгоритм. Из сказанного очевиден алгоритм сложения моментов. Мы начинаем с единственного состояния с $m = j_1 + j_2$, а именно: $|j_1, j_1; j_2, j_2\rangle$. Оно имеет полный момент $j = j_1 + j_2$, т. е. это есть $|j_1 + j_2, j_1 + j_2\rangle$. Последовательно действуя на него понижающим оператором $\hat{f}_- = \hat{f}_{1-} + \hat{f}_{2-}$ (и деля на нормировочный коэффициент согласно (4)), мы построим все состояния с этим моментом: $|j_1 + j_2, j_1 + j_2 - 1\rangle, \dots, |j_1 + j_2, -(j_1 + j_2)\rangle$. Далее переходим к проекции $m = j_1 + j_2 - 1$, и выбираем состояние, ортогональное к уже найденному состоянию с моментом $j = j_1 + j_2$. Оно имеет $j = j_1 + j_2 - 1$, т. е. это $|j_1 + j_2 - 1, j_1 + j_2 - 1\rangle$. Из него при помощи понижающего оператора строим все остальные состояния с $j = j_1 + j_2 - 1$.

Дальше действуем таким же образом. В начале каждого шага, при рассмотрении нового значения проекции m , необходимо построить состояние, ортогональное ко всем ранее полученным состояниям с тем же m . Это делается так: начинаем с произвольного состояния, например $|j_1, j_1; j_2, m - j_1\rangle$, вычитаем из него компоненты, параллельные этим ранее полученным состояниям, и затем нормируем результат. Далее из этого состояния действием \hat{f}_- получаем остальные состояния с тем же моментом.

Программа. Здесь приведена процедура ADDJ сложения двух моментов J_1 и J_2 и результаты ее работы в нескольких случаях. Поскольку все j и m могут быть целыми или полуцелыми, а циклы в REDUCE должны иметь целый индекс, в программе везде используются их удвоенные значения.

Для получения результатов достаточно было бы распечатывать значения функции $FF(J, M)$ по мере их получения. Но такая распечатка была бы не очень удобна для чтения. Во-первых, нужно в уме делить пополам все J и M в аргументах FF и F . Во-вторых, REDUCE записывает выражения с корнями из целых чисел не всегда самым удобным образом. Поэтому в процедуре ADDJ производится распечатка результатов в удобных обозначениях. Коэффициенты записываются в виде корней из рациональных чисел.

```
% ADDING MOMENTA J1 AND J2 ;
PROCEDURE ADDJ(J1, J2);
BEGIN SCALAR G, H;
```

```
J1:=2*J1;J2:=2*J2;
IF NOT FIXP J1 OR NOT FIXP J2 OR J1<0 OR J2<0 THEN
  << WRITE "ILLEGAL MOMENTUM";RETURN; >>;
IF J1<J2 THEN BEGIN SCALAR J;J:=J1;J1:=J2;J2:=J END;
OPERATOR F,FF;ON DIV;
% F(M1,M2) = !J1/2,M1/2;J2/2,M2/2 >;
% FF(J,M) = !J/2,M/2 >;
FOR ALL M1,M2 LET F(M1,M2)**2=1;
FOR ALL M1,M2,N1,N2 LET F(M1,M2)*F(N1,N2)=0;
% JM— LOWERING OPERATOR;
FOR ALL M1,M2 LET JM*F(M1,M2)=
  ((J1+M1)*(J1-M1+2))**(1/2)*F(M1-2,M2)+
  ((J2+M2)*(J2-M2+2))**(1/2)*F(M1,M2-2);
FOR J:=J1+J2 STEP -2 UNTIL J1-J2 DO
  << G:=F(J1,J-J1);G:=G-
    FOR JJ:=J1+J2 STEP -2 UNTIL J+2 SUM
      << H:=G*FF(JJ,J);H*FF(JJ,J) >>;
    H:=G*G;FF(J,J):=G/H**(1/2);
    FOR M:=J-2 STEP -2 UNTIL -J DO
      FF(J,M):=JM*FF(J,M+2)/((J+M+2)*(J-M))**(1/2)
  >>;
% PRINTING OF THE RESULTS ;
FOR J:=J1+J2 STEP -2 UNTIL J1-J2 DO
  FOR M:=J STEP -2 UNTIL -J DO
    << WRITE "!","J/2,","M/2,")="";
    FOR M1:=J1 STEP -2 UNTIL -J1 DO
      << G:=FF(J,M)*F(M1,M-M1);
        IF G NEQ 0 THEN
          << H:=G**2;G:=G/H**(1/2);
            IF H=1
              THEN WRITE " ",
                IF G=-1 THEN "- " ELSE "+ ","!",
                J1/2,","M1/2,","J2/2,","(M-M1)/2,">"
            ELSE WRITE " ",
                IF G=-1 THEN "- " ELSE "+ ",
                " SQRT(","H,") !",
                J1/2,","M1/2,","J2/2,","(M-M1)/2,">"
          >>
        >>
      >>
    >>;
  CLEAR F,FF;
```

END;
 ADDJ
 ADDJ(1/2,1/2)\$
 $|1,1\rangle =$
 $+|1/2,1/2;1/2,1/2\rangle$
 $|1,0\rangle =$
 $+ \text{SQRT}(1/2) |1/2,1/2;1/2,-1/2\rangle$
 $+ \text{SQRT}(1/2) |1/2,-1/2;1/2,1/2\rangle$
 $|1,(-1)\rangle =$
 $+|1/2,-1/2;1/2,-1/2\rangle$
 $|0,0\rangle =$
 $+ \text{SQRT}(1/2) |1/2,1/2;1/2,-1/2\rangle$
 $- \text{SQRT}(1/2) |1/2,-1/2;1/2,1/2\rangle$
 ADDJ(1/2,1)\$
 $|3/2,3/2\rangle =$
 $+|1,1;1/2,1/2\rangle$
 $|3/2,1/2\rangle =$
 $+ \text{SQRT}(1/3) |1,1;1/2,-1/2\rangle$
 $+ \text{SQRT}(2/3) |1,0;1/2,1/2\rangle$
 $|3/2,-1/2\rangle =$
 $+ \text{SQRT}(2/3) |1,0;1/2,-1/2\rangle$
 $+ \text{SQRT}(1/3) |1,(-1);1/2,1/2\rangle$
 $|3/2,-3/2\rangle =$
 $+|1,(-1);1/2,-1/2\rangle$
 $|1/2,1/2\rangle =$
 $+ \text{SQRT}(2/3) |1,1;1/2,-1/2\rangle$
 $- \text{SQRT}(1/3) |1,0;1/2,1/2\rangle$
 $|1/2,-1/2\rangle =$
 $+ \text{SQRT}(1/3) |1,0;1/2,-1/2\rangle$
 $- \text{SQRT}(2/3) |1,(-1);1/2,1/2\rangle$
 ADDJ(1/2,3/2)\$
 $|2,2\rangle =$
 $+|3/2,3/2;1/2,1/2\rangle$
 $|2,1\rangle =$
 $+ \text{SQRT}(1/4) |3/2,3/2;1/2,-1/2\rangle$
 $+ \text{SQRT}(3/4) |3/2,1/2;1/2,1/2\rangle$
 $|2,0\rangle =$
 $+ \text{SQRT}(1/2) |3/2,1/2;1/2,-1/2\rangle$

$+ \text{SQRT}(1/2) |3/2,-1/2;1/2,1/2\rangle$
 $|2,(-1)\rangle =$
 $+ \text{SQRT}(3/4) |3/2,-1/2;1/2,-1/2\rangle$
 $+ \text{SQRT}(1/4) |3/2,-3/2;1/2,1/2\rangle$
 $|2,(-2)\rangle =$
 $+|3/2,-3/2;1/2,-1/2\rangle$
 $|1,1\rangle =$
 $+ \text{SQRT}(3/4) |3/2,3/2;1/2,-1/2\rangle$
 $+ \text{SQRT}(1/4) |3/2,1/2;1/2,1/2\rangle$
 $|1,0\rangle =$
 $+ \text{SQRT}(1/2) |3/2,1/2;1/2,-1/2\rangle$
 $+ \text{SQRT}(1/2) |3/2,-1/2;1/2,1/2\rangle$
 $|1,(-1)\rangle =$
 $+ \text{SQRT}(1/4) |3/2,-1/2;1/2,-1/2\rangle$
 $- \text{SQRT}(3/4) |3/2,-3/2;1/2,1/2\rangle$

ADDJ(1/2,2)\$
 $|5/2,5/2\rangle =$
 $+|2,2;1/2,1/2\rangle$
 $|5/2,3/2\rangle =$
 $+ \text{SQRT}(1/5) |2,2;1/2,-1/2\rangle$
 $+ \text{SQRT}(4/5) |2,1;1/2,1/2\rangle$
 $|5/2,1/2\rangle =$
 $+ \text{SQRT}(2/5) |2,1;1/2,-1/2\rangle$
 $+ \text{SQRT}(3/5) |2,0;1/2,1/2\rangle$
 $|5/2,-1/2\rangle =$
 $+ \text{SQRT}(3/5) |2,0;1/2,-1/2\rangle$
 $+ \text{SQRT}(2/5) |2,(-1);1/2,1/2\rangle$
 $|5/2,-3/2\rangle =$
 $+ \text{SQRT}(4/5) |2,(-1);1/2,-1/2\rangle$
 $+ \text{SQRT}(1/5) |2,(-2);1/2,1/2\rangle$
 $|5/2,-5/2\rangle =$
 $+|2,(-2);1/2,-1/2\rangle$
 $|3/2,3/2\rangle =$
 $+ \text{SQRT}(4/5) |2,2;1/2,-1/2\rangle$
 $- \text{SQRT}(1/5) |2,1;1/2,1/2\rangle$
 $|3/2,1/2\rangle =$
 $+ \text{SQRT}(3/5) |2,1;1/2,-1/2\rangle$
 $- \text{SQRT}(2/5) |2,0;1/2,1/2\rangle$
 $|3/2,-1/2\rangle =$
 $+ \text{SQRT}(2/5) |2,0;1/2,-1/2\rangle$

$$\begin{aligned}
& + \text{SQRT}(3/5) |2, (-1); 1/2, 1/2\rangle \\
|3/2, -3/2\rangle = & \\
& + \text{SQRT}(1/5) |2, (-1); 1/2, -1/2\rangle \\
& - \text{SQRT}(4/5) |2, (-2); 1/2, 1/2\rangle
\end{aligned}$$

ADDJ(1,1)\$

$$\begin{aligned}
|2,2\rangle = & \\
& + |1,1; 1,1\rangle \\
|2,1\rangle = & \\
& + \text{SQRT}(1/2) |1,1; 1,0\rangle \\
& + \text{SQRT}(1/2) |1,0; 1,1\rangle \\
|2,0\rangle = & \\
& + \text{SQRT}(1/6) |1,1; 1, (-1)\rangle \\
& + \text{SQRT}(2/3) |1,0; 1,0\rangle \\
& + \text{SQRT}(1/6) |1, (-1); 1,1\rangle \\
|2, (-1)\rangle = & \\
& + \text{SQRT}(1/2) |1,0; 1, (-1)\rangle \\
& + \text{SQRT}(1/2) |1, (-1); 1,0\rangle \\
|2, (-2)\rangle = & \\
& + |1, (-1); 1, (-1)\rangle \\
|1,1\rangle = & \\
& + \text{SQRT}(1/2) |1,1; 1,0\rangle \\
& - \text{SQRT}(1/2) |1,0; 1,1\rangle \\
|1,0\rangle = & \\
& + \text{SQRT}(1/2) |1,1; 1, (-1)\rangle \\
& - \text{SQRT}(1/2) |1, (-1); 1,1\rangle \\
|1, (-1)\rangle = & \\
& + \text{SQRT}(1/2) |1,0; 1, (-1)\rangle \\
& - \text{SQRT}(1/2) |1, (-1); 1,0\rangle \\
|0,0\rangle = & \\
& + \text{SQRT}(1/3) |1,1; 1, (-1)\rangle \\
& - \text{SQRT}(1/3) |1,0; 1,0\rangle \\
& + \text{SQRT}(1/3) |1, (-1); 1,1\rangle
\end{aligned}$$

ADDJ(1,3/2)\$

$$\begin{aligned}
|5/2, 5/2\rangle = & \\
& + |3/2, 3/2; 1,1\rangle \\
|5/2, 3/2\rangle = & \\
& + \text{SQRT}(2/5) |3/2, 3/2; 1,0\rangle \\
& + \text{SQRT}(3/5) |3/2, 1/2; 1,1\rangle \\
|5/2, 1/2\rangle = &
\end{aligned}$$

$$\begin{aligned}
& + \text{SQRT}(1/10) |3/2, 3/2; 1, (-1)\rangle \\
& + \text{SQRT}(3/5) |3/2, 1/2; 1,0\rangle \\
& + \text{SQRT}(3/10) |3/2, -1/2; 1,1\rangle \\
|5/2, -1/2\rangle = & \\
& + \text{SQRT}(3/10) |3/2, 1/2; 1, (-1)\rangle \\
& + \text{SQRT}(3/5) |3/2, -1/2; 1,0\rangle \\
& + \text{SQRT}(1/10) |3/2, -3/2; 1,1\rangle \\
|5/2, -3/2\rangle = & \\
& + \text{SQRT}(3/5) |3/2, -1/2; 1, (-1)\rangle \\
& + \text{SQRT}(2/5) |3/2, -3/2; 1,0\rangle \\
|5/2, -5/2\rangle = & \\
& + |3/2, -3/2; 1, (-1)\rangle \\
|3/2, 3/2\rangle = & \\
& + \text{SQRT}(3/5) |3/2, 3/2; 1,0\rangle \\
& + \text{SQRT}(2/5) |3/2, 1/2; 1,1\rangle \\
|3/2, 1/2\rangle = & \\
& + \text{SQRT}(2/5) |3/2, 3/2; 1, (-1)\rangle \\
& + \text{SQRT}(1/15) |3/2, 1/2; 1,0\rangle \\
& + \text{SQRT}(8/15) |3/2, -1/2; 1,1\rangle \\
|3/2, -1/2\rangle = & \\
& + \text{SQRT}(8/15) |3/2, 1/2; 1, (-1)\rangle \\
& + \text{SQRT}(1/15) |3/2, -1/2; 1,0\rangle \\
& + \text{SQRT}(2/5) |3/2, -3/2; 1,1\rangle \\
|3/2, -3/2\rangle = & \\
& + \text{SQRT}(2/5) |3/2, -1/2; 1, (-1)\rangle \\
& - \text{SQRT}(3/5) |3/2, -3/2; 1,0\rangle \\
|1/2, 1/2\rangle = & \\
& + \text{SQRT}(1/2) |3/2, 3/2; 1, (-1)\rangle \\
& + \text{SQRT}(1/3) |3/2, 1/2; 1,0\rangle \\
& + \text{SQRT}(1/6) |3/2, -1/2; 1,1\rangle \\
|1/2, -1/2\rangle = & \\
& + \text{SQRT}(1/6) |3/2, 1/2; 1, (-1)\rangle \\
& + \text{SQRT}(1/3) |3/2, -1/2; 1,0\rangle \\
& + \text{SQRT}(1/2) |3/2, -3/2; 1,1\rangle
\end{aligned}$$

ADDJ(1,2)\$

$$\begin{aligned}
|3,3\rangle = & \\
& + |2,2; 1,1\rangle \\
|3,2\rangle = & \\
& + \text{SQRT}(1/3) |2,2; 1,0\rangle \\
& + \text{SQRT}(2/3) |2,1; 1,1\rangle
\end{aligned}$$

$$\begin{aligned}
|3,1\rangle &= \\
&+ \text{SQRT}(1/15) |2,2;1,(-1)\rangle \\
&+ \text{SQRT}(8/15) |2,1;1,0\rangle \\
&+ \text{SQRT}(2/5) |2,0;1,1\rangle \\
|3,0\rangle &= \\
&+ \text{SQRT}(1/5) |2,1;1,(-1)\rangle \\
&+ \text{SQRT}(3/5) |2,0;1,0\rangle \\
&+ \text{SQRT}(1/5) |2,(-1);1,1\rangle \\
|3,(-1)\rangle &= \\
&+ \text{SQRT}(2/5) |2,0;1,(-1)\rangle \\
&+ \text{SQRT}(8/15) |2,(-1);1,0\rangle \\
&+ \text{SQRT}(1/15) |2,(-2);1,1\rangle \\
|3,(-2)\rangle &= \\
&+ \text{SQRT}(2/3) |2,(-1);1,(-1)\rangle \\
&+ \text{SQRT}(1/3) |2,(-2);1,0\rangle \\
|3,(-3)\rangle &= \\
&+ |2,(-2);1,(-1)\rangle \\
|2,2\rangle &= \\
&+ \text{SQRT}(2/3) |2,2;1,0\rangle \\
&- \text{SQRT}(1/3) |2,1;1,1\rangle \\
|2,1\rangle &= \\
&+ \text{SQRT}(1/3) |2,2;1,(-1)\rangle \\
&+ \text{SQRT}(1/6) |2,1;1,0\rangle \\
&+ \text{SQRT}(1/2) |2,0;1,1\rangle \\
|2,0\rangle &= \\
&+ \text{SQRT}(1/2) |2,1;1,(-1)\rangle \\
&+ \text{SQRT}(1/2) |2,(-1);1,1\rangle \\
|2,(-1)\rangle &= \\
&+ \text{SQRT}(1/2) |2,0;1,(-1)\rangle \\
&+ \text{SQRT}(1/6) |2,(-1);1,0\rangle \\
&- \text{SQRT}(1/3) |2,(-2);1,1\rangle \\
|2,(-2)\rangle &= \\
&+ \text{SQRT}(1/3) |2,(-1);1,(-1)\rangle \\
&- \text{SQRT}(2/3) |2,(-2);1,0\rangle \\
|1,1\rangle &= \\
&+ \text{SQRT}(3/5) |2,2;1,(-1)\rangle \\
&+ \text{SQRT}(3/10) |2,1;1,0\rangle \\
&+ \text{SQRT}(1/10) |2,0;1,1\rangle \\
|1,0\rangle &= \\
&+ \text{SQRT}(3/10) |2,1;1,(-1)\rangle \\
&+ \text{SQRT}(2/5) |2,0;1,0\rangle
\end{aligned}$$

$$\begin{aligned}
&+ \text{SQRT}(3/10) |2,(-1);1,1\rangle \\
|1,(-1)\rangle &= \\
&+ \text{SQRT}(1/10) |2,0;1,(-1)\rangle \\
&+ \text{SQRT}(3/10) |2,(-1);1,0\rangle \\
&+ \text{SQRT}(3/5) |2,(-2);1,1\rangle \\
\text{ADDJ}(3/2,3/2) \$ \\
|3,3\rangle &= \\
&+ |3/2,3/2;3/2,3/2\rangle \\
|3,2\rangle &= \\
&+ \text{SQRT}(1/2) |3/2,3/2;3/2,1/2\rangle \\
&+ \text{SQRT}(1/2) |3/2,1/2;3/2,3/2\rangle \\
|3,1\rangle &= \\
&+ \text{SQRT}(1/5) |3/2,3/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(3/5) |3/2,1/2;3/2,1/2\rangle \\
&+ \text{SQRT}(1/5) |3/2,-1/2;3/2,3/2\rangle \\
|3,0\rangle &= \\
&+ \text{SQRT}(1/20) |3/2,3/2;3/2,-3/2\rangle \\
&+ \text{SQRT}(9/20) |3/2,1/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(9/20) |3/2,-1/2;3/2,1/2\rangle \\
&+ \text{SQRT}(1/20) |3/2,-3/2;3/2,3/2\rangle \\
|3,(-1)\rangle &= \\
&+ \text{SQRT}(1/5) |3/2,1/2;3/2,-3/2\rangle \\
&+ \text{SQRT}(3/5) |3/2,-1/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(1/5) |3/2,-3/2;3/2,1/2\rangle \\
|3,(-2)\rangle &= \\
&+ \text{SQRT}(1/2) |3/2,-1/2;3/2,-3/2\rangle \\
&+ \text{SQRT}(1/2) |3/2,-3/2;3/2,-1/2\rangle \\
|3,(-3)\rangle &= \\
&+ |3/2,-3/2;3/2,-3/2\rangle \\
|2,2\rangle &= \\
&+ \text{SQRT}(1/2) |3/2,3/2;3/2,1/2\rangle \\
&- \text{SQRT}(1/2) |3/2,1/2;3/2,3/2\rangle \\
|2,1\rangle &= \\
&+ \text{SQRT}(1/2) |3/2,3/2;3/2,-1/2\rangle \\
&- \text{SQRT}(1/2) |3/2,-1/2;3/2,3/2\rangle \\
|2,0\rangle &= \\
&+ \text{SQRT}(1/4) |3/2,3/2;3/2,-3/2\rangle \\
&+ \text{SQRT}(1/4) |3/2,1/2;3/2,-1/2\rangle \\
&- \text{SQRT}(1/4) |3/2,-1/2;3/2,1/2\rangle \\
&- \text{SQRT}(1/4) |3/2,-3/2;3/2,3/2\rangle
\end{aligned}$$

$$\begin{aligned}
|2,(-1)\rangle &= \\
&+ \text{SQRT}(1/2) |3/2,1/2;3/2,-3/2\rangle \\
&- \text{SQRT}(1/2) |3/2,-3/2;3/2,1/2\rangle \\
|2,(-2)\rangle &= \\
&+ \text{SQRT}(1/2) |3/2,-1/2;3/2,-3/2\rangle \\
&- \text{SQRT}(1/2) |3/2,-3/2;3/2,-1/2\rangle \\
|1,1\rangle &= \\
&+ \text{SQRT}(3/10) |3/2,3/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(2/5) |3/2,1/2;3/2,1/2\rangle \\
&+ \text{SQRT}(3/10) |3/2,-1/2;3/2,3/2\rangle \\
|1,0\rangle &= \\
&+ \text{SQRT}(9/20) |3/2,3/2;3/2,-3/2\rangle \\
&+ \text{SQRT}(1/20) |3/2,1/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(1/20) |3/2,-1/2;3/2,1/2\rangle \\
&+ \text{SQRT}(9/20) |3/2,-3/2;3/2,3/2\rangle \\
|1,(-1)\rangle &= \\
&+ \text{SQRT}(3/10) |3/2,1/2;3/2,-3/2\rangle \\
&+ \text{SQRT}(2/5) |3/2,-1/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(3/10) |3/2,-3/2;3/2,1/2\rangle \\
|0,0\rangle &= \\
&+ \text{SQRT}(1/4) |3/2,3/2;3/2,-3/2\rangle \\
&- \text{SQRT}(1/4) |3/2,1/2;3/2,-1/2\rangle \\
&+ \text{SQRT}(1/4) |3/2,-1/2;3/2,1/2\rangle \\
&- \text{SQRT}(1/4) |3/2,-3/2;3/2,3/2\rangle \\
\text{ADDJ}(3/2,2)\$ \\
|7/2,7/2\rangle &= \\
&+ |2,2;3/2,3/2\rangle \\
|7/2,5/2\rangle &= \\
&+ \text{SQRT}(3/7) |2,2;3/2,1/2\rangle \\
&+ \text{SQRT}(4/7) |2,1;3/2,3/2\rangle \\
|7/2,3/2\rangle &= \\
&+ \text{SQRT}(1/7) |2,2;3/2,-1/2\rangle \\
&+ \text{SQRT}(4/7) |2,1;3/2,1/2\rangle \\
&+ \text{SQRT}(2/7) |2,0;3/2,3/2\rangle \\
|7/2,1/2\rangle &= \\
&+ \text{SQRT}(1/35) |2,2;3/2,-3/2\rangle \\
&+ \text{SQRT}(12/35) |2,1;3/2,-1/2\rangle \\
&+ \text{SQRT}(18/35) |2,0;3/2,1/2\rangle \\
&+ \text{SQRT}(4/35) |2,(-1);3/2,3/2\rangle \\
|7/2,-1/2\rangle &=
\end{aligned}$$

$$\begin{aligned}
&+ \text{SQRT}(4/35) |2,1;3/2,-3/2\rangle \\
&+ \text{SQRT}(18/35) |2,0;3/2,-1/2\rangle \\
&+ \text{SQRT}(12/35) |2,(-1);3/2,1/2\rangle \\
&+ \text{SQRT}(1/35) |2,(-2);3/2,3/2\rangle \\
|7/2,-3/2\rangle &= \\
&+ \text{SQRT}(2/7) |2,0;3/2,-3/2\rangle \\
&+ \text{SQRT}(4/7) |2,(-1);3/2,-1/2\rangle \\
&+ \text{SQRT}(1/7) |2,(-2);3/2,1/2\rangle \\
|7/2,-5/2\rangle &= \\
&+ \text{SQRT}(4/7) |2,(-1);3/2,-3/2\rangle \\
&+ \text{SQRT}(3/7) |2,(-2);3/2,-1/2\rangle \\
|7/2,\pm 7/2\rangle &= \\
&+ |2,(-2);3/2,-3/2\rangle \\
|5/2,5/2\rangle &= \\
&+ \text{SQRT}(4/7) |2,2;3/2,1/2\rangle \\
&+ \text{SQRT}(3/7) |2,1;3/2,3/2\rangle \\
|5/2,3/2\rangle &= \\
&+ \text{SQRT}(16/35) |2,2;3/2,-1/2\rangle \\
&+ \text{SQRT}(1/35) |2,1;3/2,1/2\rangle \\
&+ \text{SQRT}(18/35) |2,0;3/2,3/2\rangle \\
|5/2,1/2\rangle &= \\
&+ \text{SQRT}(6/35) |2,2;3/2,\pm 3/2\rangle \\
&+ \text{SQRT}(5/14) |2,1;3/2,-1/2\rangle \\
&+ \text{SQRT}(3/35) |2,0;3/2,1/2\rangle \\
&+ \text{SQRT}(27/70) |2,(-1);3/2,3/2\rangle \\
|5/2,-1/2\rangle &= \\
&+ \text{SQRT}(27/70) |2,1;3/2,-3/2\rangle \\
&+ \text{SQRT}(3/35) |2,0;3/2,-1/2\rangle \\
&+ \text{SQRT}(5/14) |2,(-1);3/2,1/2\rangle \\
&+ \text{SQRT}(6/35) |2,(-2);3/2,3/2\rangle \\
|5/2,-3/2\rangle &= \\
&+ \text{SQRT}(18/35) |2,0;3/2,-3/2\rangle \\
&+ \text{SQRT}(1/35) |2,(-1);3/2,-1/2\rangle \\
&+ \text{SQRT}(16/35) |2,(-2);3/2,1/2\rangle \\
|5/2,-5/2\rangle &= \\
&+ \text{SQRT}(4/5) |2,2;1/2,-1/2\rangle \\
&- \text{SQRT}(1/5) |2,1;1/2,1/2\rangle \\
|3/2,1/2\rangle &= \\
&+ \text{SQRT}(3/5) |2,1;1/2,-1/2\rangle \\
&- \text{SQRT}(2/5) |2,0;1/2,1/2\rangle
\end{aligned}$$

$$\begin{aligned}
|3/2, 1/2\rangle = & \\
& + \text{SQRT}(2/5) |2, 2; 3/2, -3/2\rangle \\
& - \text{SQRT}(1/5) |2, 0; 3/2, 1/2\rangle \\
& + \text{SQRT}(2/5) |2, (-1); 3/2, 3/2\rangle
\end{aligned}$$

$$\begin{aligned}
|3/2, -1/2\rangle = & \\
& + \text{SQRT}(2/5) |2, 1; 3/2, -3/2\rangle \\
& - \text{SQRT}(1/5) |2, 0; 3/2, -1/2\rangle \\
& + \text{SQRT}(2/5) |2, (-2); 3/2, 3/2\rangle
\end{aligned}$$

$$\begin{aligned}
|3/2, -3/2\rangle = & \\
& + \text{SQRT}(1/5) |2, 0; 3/2, -3/2\rangle \\
& - \text{SQRT}(2/5) |2, (-1); 3/2, -1/2\rangle \\
& + \text{SQRT}(2/5) |2, (-2); 3/2, 1/2\rangle
\end{aligned}$$

$$\begin{aligned}
|1/2, 1/2\rangle = & \\
& + \text{SQRT}(2/5) |2, 2; 3/2, -3/2\rangle \\
& + \text{SQRT}(3/10) |2, 1; 3/2, -1/2\rangle \\
& + \text{SQRT}(1/5) |2, 0; 3/2, 1/2\rangle \\
& + \text{SQRT}(1/10) |2, (-1); 3/2, 3/2\rangle
\end{aligned}$$

$$\begin{aligned}
|1/2, -1/2\rangle = & \\
& + \text{SQRT}(1/10) |2, 1; 3/2, -3/2\rangle \\
& - \text{SQRT}(1/5) |2, 0; 3/2, -1/2\rangle \\
& + \text{SQRT}(3/10) |2, (-1); 3/2, 1/2\rangle \\
& - \text{SQRT}(2/5) |2, (-2); 3/2, 3/2\rangle
\end{aligned}$$

ADDJ(2,2) \$

$$\begin{aligned}
|4, 4\rangle = & \\
& + |2, 2; 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|4, 3\rangle = & \\
& + \text{SQRT}(1/2) |2, 2; 2, 1\rangle \\
& + \text{SQRT}(1/2) |2, 1; 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|4, 2\rangle = & \\
& + \text{SQRT}(3/14) |2, 2; 2, 0\rangle \\
& + \text{SQRT}(4/7) |2, 1; 2, 1\rangle \\
& + \text{SQRT}(3/14) |2, 0; 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|4, 1\rangle = & \\
& + \text{SQRT}(1/14) |2, 2; 2, (-1)\rangle \\
& + \text{SQRT}(3/7) |2, 1; 2, 0\rangle \\
& + \text{SQRT}(3/7) |2, 0; 2, 1\rangle \\
& + \text{SQRT}(1/14) |2, (-1); 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|4, 0\rangle = & \\
& + \text{SQRT}(1/70) |2, 2; 2, (-2)\rangle \\
& + \text{SQRT}(8/35) |2, 1; 2, (-1)\rangle
\end{aligned}$$

$$\begin{aligned}
& + \text{SQRT}(18/35) |2, 0; 2, 0\rangle \\
& + \text{SQRT}(8/35) |2, (-1); 2, 1\rangle \\
& + \text{SQRT}(1/70) |2, (-2); 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|4, (-1)\rangle = & \\
& + \text{SQRT}(1/14) |2, 1; 2, (-2)\rangle \\
& + \text{SQRT}(3/7) |2, 0; 2, (-1)\rangle \\
& + \text{SQRT}(3/7) |2, (-1); 2, 0\rangle \\
& + \text{SQRT}(1/14) |2, (-2); 2, 1\rangle
\end{aligned}$$

$$\begin{aligned}
|4, (-2)\rangle = & \\
& + \text{SQRT}(3/14) |2, 0; 2, (-2)\rangle \\
& + \text{SQRT}(4/7) |2, (-1); 2, (-1)\rangle \\
& + \text{SQRT}(3/14) |2, (-2); 2, 0\rangle
\end{aligned}$$

$$\begin{aligned}
|4, (-3)\rangle = & \\
& + \text{SQRT}(1/2) |2, (-1); 2, (-2)\rangle \\
& + \text{SQRT}(1/2) |2, (-2); 2, (-1)\rangle
\end{aligned}$$

$$\begin{aligned}
|4, (-4)\rangle = & \\
& + |2, (-2); 2, (-2)\rangle
\end{aligned}$$

$$\begin{aligned}
|3, 3\rangle = & \\
& + \text{SQRT}(1/2) |2, 2; 2, 1\rangle \\
& - \text{SQRT}(1/2) |2, 1; 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|3, 2\rangle = & \\
& + \text{SQRT}(1/2) |2, 2; 2, 0\rangle \\
& + \text{SQRT}(1/2) |2, 0; 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|3, 1\rangle = & \\
& + \text{SQRT}(3/10) |2, 2; 2, (-1)\rangle \\
& + \text{SQRT}(1/5) |2, 1; 2, 0\rangle \\
& + \text{SQRT}(1/5) |2, 0; 2, 1\rangle \\
& - \text{SQRT}(3/10) |2, (-1); 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|3, 0\rangle = & \\
& + \text{SQRT}(1/10) |2, 2; 2, (-2)\rangle \\
& + \text{SQRT}(2/5) |2, 1; 2, (-1)\rangle \\
& + \text{SQRT}(2/5) |2, (-1); 2, 1\rangle \\
& + \text{SQRT}(1/10) |2, (-2); 2, 2\rangle
\end{aligned}$$

$$\begin{aligned}
|3, (-1)\rangle = & \\
& + \text{SQRT}(3/10) |2, 1; 2, (-2)\rangle \\
& + \text{SQRT}(1/5) |2, 0; 2, (-1)\rangle \\
& + \text{SQRT}(1/5) |2, (-1); 2, 0\rangle \\
& - \text{SQRT}(3/10) |2, (-2); 2, 1\rangle
\end{aligned}$$

$$\begin{aligned}
|3, (-2)\rangle = & \\
& + \text{SQRT}(1/2) |2, 0; 2, (-2)\rangle \\
& + \text{SQRT}(1/2) |2, (-2); 2, 0\rangle
\end{aligned}$$

$$\begin{aligned}
|3,(-3)\rangle &= \\
&+ \text{SQRT}(1/2) |2,(-1);2,(-2)\rangle \\
&- \text{SQRT}(1/2) |2,(-2);2,(-1)\rangle \\
|2,2\rangle &= \\
&+ \text{SQRT}(2/7) |2,2;2,0\rangle \\
&+ \text{SQRT}(3/7) |2,1;2,1\rangle \\
&+ \text{SQRT}(2/7) |2,0;2,2\rangle \\
|2,1\rangle &= \\
&+ \text{SQRT}(3/7) |2,2;2,(-1)\rangle \\
&+ \text{SQRT}(1/14) |2,1;2,0\rangle \\
&+ \text{SQRT}(1/14) |2,0;2,1\rangle \\
&+ \text{SQRT}(3/7) |2,(-1);2,2\rangle \\
|2,0\rangle &= \\
&+ \text{SQRT}(2/7) |2,2;2,(-2)\rangle \\
&+ \text{SQRT}(1/14) |2,1;2,(-1)\rangle \\
&- \text{SQRT}(2/7) |2,0;2,0\rangle \\
&+ \text{SQRT}(1/14) |2,(-1);2,1\rangle \\
&+ \text{SQRT}(2/7) |2,(-2);2,2\rangle \\
|2,(-1)\rangle &= \\
&+ \text{SQRT}(3/7) |2,1;2,(-2)\rangle \\
&+ \text{SQRT}(1/14) |2,0;2,(-1)\rangle \\
&+ \text{SQRT}(1/14) |2,(-1);2,0\rangle \\
&+ \text{SQRT}(3/7) |2,(-2);2,1\rangle \\
|2,(-2)\rangle &= \\
&+ \text{SQRT}(2/7) |2,0;2,(-2)\rangle \\
&+ \text{SQRT}(3/7) |2,(-1);2,(-1)\rangle \\
&+ \text{SQRT}(2/7) |2,(-2);2,0\rangle \\
|1,1\rangle &= \\
&+ \text{SQRT}(1/5) |2,2;2,(-1)\rangle \\
&+ \text{SQRT}(3/10) |2,1;2,0\rangle \\
&+ \text{SQRT}(3/10) |2,0;2,1\rangle \\
&- \text{SQRT}(1/5) |2,(-1);2,2\rangle \\
|1,0\rangle &= \\
&+ \text{SQRT}(2/5) |2,2;2,(-2)\rangle \\
&+ \text{SQRT}(1/10) |2,1;2,(-1)\rangle \\
&+ \text{SQRT}(1/10) |2,(-1);2,1\rangle \\
&- \text{SQRT}(2/5) |2,(-2);2,2\rangle \\
|1,(-1)\rangle &= \\
&+ \text{SQRT}(1/5) |2,1;2,(-2)\rangle \\
&+ \text{SQRT}(3/10) |2,0;2,(-1)\rangle \\
&+ \text{SQRT}(3/10) |2,(-1);2,0\rangle
\end{aligned}$$

$$\begin{aligned}
&- \text{SQRT}(1/5) |2,(-2);2,1\rangle \\
|0,0\rangle &= \\
&+ \text{SQRT}(1/5) |2,2;2,(-2)\rangle \\
&- \text{SQRT}(1/5) |2,1;2,(-1)\rangle \\
&+ \text{SQRT}(1/5) |2,0;2,0\rangle \\
&- \text{SQRT}(1/5) |2,(-1);2,1\rangle \\
&+ \text{SQRT}(1/5) |2,(-2);2,2\rangle
\end{aligned}$$

ЛИТЕРАТУРА

1. Ландау Л.Д., Лифшиц Е.М. Гидродинамика. — М.: Наука, 1986.
2. Сретенский Л.Н. Теория волновых движений жидкости. — М.: Наука, 1977.
3. Gerdt V.P., Kostov N.A., Shvachka A.V. Preprint JINR E11-83-750, 1983.
4. Ландау Л.Д., Лифшиц Е.М. Теория поля. — М.: Наука, 1973.
5. Мизнер Ч., Торн К., Уилер Дж. Гравитация, т.1. — М.: Мир, 1977.
6. Рашевский П.К. Риманова геометрия и тензорный анализ. — М.: Наука, 1967.
7. Пенроуз Р., Риндлер В. Спиноры и пространство-время. — М.: Мир, 1987.
8. Ландау Л.Д., Лифшиц Е.М. Квантовая механика. — М.: Наука, 1974.
9. Елютин П.В., Кривченков В.Д. Квантовая механика. — М.: Наука, 1976.

А.Г. Грозин

Решение физических задач на языке REDUCE.

3. Нелинейные волны на воде

4. Вычисление тензора кривизны

5. Сложение моментов в квантовой механике

Ответственный за выпуск С.Г.Попов

Работа поступила 22 сентября 1988 г.

Подписано в печать 14.10.88 г. МН 08557

Формат бумаги 60×90 1/16 Объем 4,9 печ.л., 3,0 уч.-изд.л.

Тираж 290 экз. Бесплатно. Заказ № 136

*Набрано в автоматизированной системе на базе фото-
наборного автомата ФА1000 и ЭВМ «Электроника» и
отпечатано на ротапринтере Института ядерной физики
СО АН СССР,*

Новосибирск, 630090, пр. академика Лаврентьева, 11.